
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 6.4.3 which is part of the LTE/SAE test suite in the ‘iwd-EUTRA-B2013-03_D14wk10’ ATS delivery.

The test case can be demonstrated to run with one vendor’s LTE UEs on two different Test Platforms (see section 5). Execution log are provided as evidence.

3. Verification Test Summary

Test Case:
TC_6_4_3
Test Group:
LTE_IRAT\6_4\Hybrid_CellsUG.ttcn
ATS Version:
iwd-EUTRA-B2013-03_D14wk10
System Simulator used:
R&S CMW500 




+


Anite Platform





UE used:
Qualcomm MSM 8974  


+ 


Qualcomm 9X25
Verification Status:
PASS


4. Corrections required for test case 6.4.3
Change 1 – Correction to function f_TC_6_4_3_EUTRA
	Function name
	f_TC_6_4_3_EUTRA()

	Reason for change
	Issue with coordination messages between EUTRA and UTRAN components:

1.   Unnecessary co-ord messages at the beginning of the TC for USIM initialization.
2.   TTCN should wait for UE to camp on cell 5 before changing cell power according to T2.

	Summary of change
	Coordination triggers amended:

1.   Removed the co-ord message.
2.   Sending to and waiting for co-ord message from the UTRAN PTC before ttcn changes the cell power to T2.

	TTCN module
	LTE_IRAT/6_4/Hybrid_CellsUG.ttcn

	MCC160 Comment
	


Before change

...

  //=========================================================================

  /*

   * @desc      REFERENCE TS 36.523-1 clause 6.4.3

   * @status

   */

  function f_TC_6_4_3_EUTRA ( ) runs on EUTRA_PTC

  // Inter-RAT cell reselection / From E-UTRA RRC_IDLE non-CSG cell to UTRA_Idle member hybrid cell

  {

    var template (value) CellPowerList_Type v_CellPowerList_T2, v_CellPowerList_T3,v_CellPowerList_T4;

    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;

    var IRAT_CoOrd_SysInfo_Type v_UTRANSysInfo;

    var CellReselectionPriority v_CellReselectionPriority_Cell5_Cell7 := 3;

    var boolean v_CSInOtherRAT := false;

    v_CellPowerList_T2 := {

      cs_CellPower (eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_T3 := {

      cs_CellPower (eutra_Cell1, tsc_ServingCellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)

    };

    v_CellPowerList_T4 := {

      cs_CellPower (eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE)

    };

    // Init

    f_EUTRA_Init(c4);

    // Set TAC

    f_EUTRA_CellInfo_SetTAC (eutra_Cell2, tsc_IdleMode_TAC_Cell2);

    // Coordinate with UTRAN component to receive UTRAN frequencies

    v_UTRANSysInfo := f_IRAT_WaitForCoOrd_SysInfo (UTRAN);

    // Send the cell info to the UTRAN component

    f_IRAT_Send_EUTRA_SysInfoToOtherPort(UTRAN, f_EUTRA_CellInfo_GetEARFCN_DL(eutra_Cell1), f_ConvertDL_BandwidthToInteger (f_EUTRA_CellInfo_GetDL_ChBandwidth (eutra_Cell1)));

    // Get FDD or TDD mode

    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);

    // Set SysInfo

    if (v_EUTRA_FDD_TDD_Mode == FDD){

      f_EUTRA_CellInfo_SetSIB6 (eutra_Cell1,

                                cs_508_SystemInformationBlockType6_Def ({ cs_CarrierFreqUTRA_FDD_CRPrio (v_UTRANSysInfo.Utran[0].FDD.Freq,

                                                                                                         v_CellReselectionPriority_Cell5_Cell7)}, omit));

      f_EUTRA_CellInfo_SetSIB6 (eutra_Cell2,

                                cs_508_SystemInformationBlockType6_Def ({ cs_CarrierFreqUTRA_FDD_CRPrio (v_UTRANSysInfo.Utran[0].FDD.Freq,

                                                                                                         v_CellReselectionPriority_Cell5_Cell7)}, omit));

    }

    else {

      f_EUTRA_CellInfo_SetSIB6 (eutra_Cell1,

                                cs_508_SystemInformationBlockType6_Def (omit,

                                                                        { cs_CarrierFreqUTRA_TDD_CRPrio (v_UTRANSysInfo.Utran[0].TDD.Freq,

                                                                                                         v_CellReselectionPriority_Cell5_Cell7)}));

      f_EUTRA_CellInfo_SetSIB6 (eutra_Cell2,

                                cs_508_SystemInformationBlockType6_Def (omit,

                                                                        { cs_CarrierFreqUTRA_TDD_CRPrio (v_UTRANSysInfo.Utran[0].TDD.Freq,

                                                                                                         v_CellReselectionPriority_Cell5_Cell7)}));

    }

    // Create and configure Cells

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell2);

    // insert USIM and add the CSG ID of Cell 5 in CSG list in USIM

    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);
    // Preamble

    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);

    f_EUTRA_TestBody_Set (true);

    // @siclog "Step 1" siclog@

    // The SS changes Cell 5 and Cell 7 levels according to the row "T1" in table 6.4.3.3.2-2.

    f_EUTRA_SendAuthParameters(eutra_Cell1, UTRAN);
    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(UTRAN);

    // @siclog "Step 13" siclog@

    // The SS changes Cell 1, Cell 2, Cell 5 and Cell 7 levels according to the row "T2" in table 6.4.3.3.2-1 and table 6.4.3.3.2-2.

    f_EUTRA_SetCellPowerList (v_CellPowerList_T2);

    // @siclog "Step 14 - 21" siclog@

    // Steps 1 to 8 of the procedure described in TS 36.508 subclause 6.4.2.7A is performed on Cell 2.

    // NOTE: The UE performs a TAU procedure and the RRC connection is released.

    f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT (eutra_Cell2, v_CSInOtherRAT, false, true, omit); // @sic R5-123734 sic@

    // @siclog "Step 22" siclog@

    // The SS changes Cell 1, Cell 2 and Cell 7 levels according to the row "T3" in table 6.4.3.3.2-1 and table 6.4.3.3.2-2.

    f_EUTRA_SetCellPowerList (v_CellPowerList_T3);

    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);

    // @siclog "Step 23-28" siclog@

    // Steps 1 to 6 of the procedure described in TS 36.508 subclause 6.4.2.7 is performed on Cell 1.

    // NOTE: The UE performs a TAU procedure and the RRC connection is released.

    f_EUTRA_TrackingAreaUpdate (eutra_Cell1);

    // @siclog "Step 29" siclog@

    // The SS changes Cell 2 and Cell 5 levels according to the row "T4" in table 6.4.3.3.2-1 and table 6.4.3.3.2-2.

    f_EUTRA_SetCellPowerList (v_CellPowerList_T4);

    f_EUTRA_SendAuthParameters(eutra_Cell1, UTRAN);

    f_EUTRA_TestBody_Set (false);

    // Test case finished, so take cell down

    f_IRAT_WaitForCoOrd_Trigger(UTRAN);

    // Release non serving cell(s)

    f_EUTRA_ReleaseAllCells ();

  } // end of f_TC_6_4_3_EUTRA

...

After change

...

  //=========================================================================

  /*

   * @desc      REFERENCE TS 36.523-1 clause 6.4.3

   * @status

   */

  function f_TC_6_4_3_EUTRA ( ) runs on EUTRA_PTC

  // Inter-RAT cell reselection / From E-UTRA RRC_IDLE non-CSG cell to UTRA_Idle member hybrid cell

  {

    var template (value) CellPowerList_Type v_CellPowerList_T2, v_CellPowerList_T3,v_CellPowerList_T4;

    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;

    var IRAT_CoOrd_SysInfo_Type v_UTRANSysInfo;

    var CellReselectionPriority v_CellReselectionPriority_Cell5_Cell7 := 3;

    var boolean v_CSInOtherRAT := false;

    v_CellPowerList_T2 := {

      cs_CellPower (eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_T3 := {

      cs_CellPower (eutra_Cell1, tsc_ServingCellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)

    };

    v_CellPowerList_T4 := {

      cs_CellPower (eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE)

    };

    // Init

    f_EUTRA_Init(c4);

    // Set TAC

    f_EUTRA_CellInfo_SetTAC (eutra_Cell2, tsc_IdleMode_TAC_Cell2);

    // Coordinate with UTRAN component to receive UTRAN frequencies

    v_UTRANSysInfo := f_IRAT_WaitForCoOrd_SysInfo (UTRAN);

    // Send the cell info to the UTRAN component

    f_IRAT_Send_EUTRA_SysInfoToOtherPort(UTRAN, f_EUTRA_CellInfo_GetEARFCN_DL(eutra_Cell1), f_ConvertDL_BandwidthToInteger (f_EUTRA_CellInfo_GetDL_ChBandwidth (eutra_Cell1)));

    // Get FDD or TDD mode

    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);

    // Set SysInfo

    if (v_EUTRA_FDD_TDD_Mode == FDD){

      f_EUTRA_CellInfo_SetSIB6 (eutra_Cell1,

                                cs_508_SystemInformationBlockType6_Def ({ cs_CarrierFreqUTRA_FDD_CRPrio (v_UTRANSysInfo.Utran[0].FDD.Freq,

                                                                                                         v_CellReselectionPriority_Cell5_Cell7)}, omit));

      f_EUTRA_CellInfo_SetSIB6 (eutra_Cell2,

                                cs_508_SystemInformationBlockType6_Def ({ cs_CarrierFreqUTRA_FDD_CRPrio (v_UTRANSysInfo.Utran[0].FDD.Freq,

                                                                                                         v_CellReselectionPriority_Cell5_Cell7)}, omit));

    }

    else {

      f_EUTRA_CellInfo_SetSIB6 (eutra_Cell1,

                                cs_508_SystemInformationBlockType6_Def (omit,

                                                                        { cs_CarrierFreqUTRA_TDD_CRPrio (v_UTRANSysInfo.Utran[0].TDD.Freq,

                                                                                                         v_CellReselectionPriority_Cell5_Cell7)}));

      f_EUTRA_CellInfo_SetSIB6 (eutra_Cell2,

                                cs_508_SystemInformationBlockType6_Def (omit,

                                                                        { cs_CarrierFreqUTRA_TDD_CRPrio (v_UTRANSysInfo.Utran[0].TDD.Freq,

                                                                                                         v_CellReselectionPriority_Cell5_Cell7)}));

    }

    // Create and configure Cells

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell2);

    // insert USIM and add the CSG ID of Cell 5 in CSG list in USIM

    //f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger); 
    // Preamble

    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);

    f_EUTRA_TestBody_Set (true);

    // @siclog "Step 1" siclog@

    // The SS changes Cell 5 and Cell 7 levels according to the row "T1" in table 6.4.3.3.2-2.

    f_EUTRA_SendAuthParameters(eutra_Cell1, UTRAN);
    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(UTRAN);

    f_EUTRA_SendTriggerToOtherRAT(true); 
    // @siclog "Step 13" siclog@

    // The SS changes Cell 1, Cell 2, Cell 5 and Cell 7 levels according to the row "T2" in table 6.4.3.3.2-1 and table 6.4.3.3.2-2.

    f_EUTRA_SetCellPowerList (v_CellPowerList_T2);

    // @siclog "Step 14 - 21" siclog@

    // Steps 1 to 8 of the procedure described in TS 36.508 subclause 6.4.2.7A is performed on Cell 2.

    // NOTE: The UE performs a TAU procedure and the RRC connection is released.

    f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT (eutra_Cell2, v_CSInOtherRAT, false, true, omit); // @sic R5-123734 sic@

    // @siclog "Step 22" siclog@

    // The SS changes Cell 1, Cell 2 and Cell 7 levels according to the row "T3" in table 6.4.3.3.2-1 and table 6.4.3.3.2-2.

    f_EUTRA_SetCellPowerList (v_CellPowerList_T3);

    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);

    // @siclog "Step 23-28" siclog@

    // Steps 1 to 6 of the procedure described in TS 36.508 subclause 6.4.2.7 is performed on Cell 1.

    // NOTE: The UE performs a TAU procedure and the RRC connection is released.

    f_EUTRA_TrackingAreaUpdate (eutra_Cell1);

    // @siclog "Step 29" siclog@

    // The SS changes Cell 2 and Cell 5 levels according to the row "T4" in table 6.4.3.3.2-1 and table 6.4.3.3.2-2.

    f_EUTRA_SetCellPowerList (v_CellPowerList_T4);

    f_EUTRA_SendAuthParameters(eutra_Cell1, UTRAN);

    f_EUTRA_TestBody_Set (false);

    // Test case finished, so take cell down

    f_IRAT_WaitForCoOrd_Trigger(UTRAN);

    // Release non serving cell(s)

    f_EUTRA_ReleaseAllCells ();

  } // end of f_TC_6_4_3_EUTRA

...

Change 2 – Correction to function f_TC_6_4_3_UTRAN
	Function name
	f_TC_6_4_3_UTRAN()

	Reason for change
	1.   Unnecessary co-ord messages at the beginning of the TC before the test body is set to true.
Moreover, TTCN should wait for UE to camp on cell 5 before changing cell power according to T2.
2.   TTCN is trying to retrieve the PLMN before the UTRAN Cell Init has been done resulting in a crash.
3.   Unnecessary Auth param sent to EUTRA PTC after camp on cell 5.
4.   When doing a reselection from LTE to UTRAN in steps 30-37, the RRC Connection Request message should have cause “Reselection” instead of “Registration”.

	Summary of change
	1.   Removed the co-ord messages as the UTRAN PTC will be waiting till the Auth params are received from EUTRA PTC after preamble is done.
Receiving and sending co-ord message from the UTRAN PTC before ttcn changes the cell power to T2.
2.   Changed the TTCN to retrieve the PLMN after the UTRAN Cell Init has been done.
3.   Removed the Auth param send to EUTRA PTC after successful camp of UE to cell 5.
4.   Corrected the used receive constraint (see also change 4 below).

	TTCN module
	LTE_IRAT/6_4/Hybrid_CellsUG_UTRAN.ttcn

	MCC160 Comment
	


Before change

...

  //========================================================================

  /*

   * @desc      REFERENCE TS 36.523-1 clause 6.4.3

   * @status

   */

  function f_TC_6_4_3_UTRAN() runs on UTRAN_PTC

  { // Inter-RAT cell reselection / From E-UTRA RRC_IDLE non-CSG cell to UTRA_Idle member hybrid cell

    var IRAT_CoOrd_SysInfo_Type v_EUTRASysInfo;

    var integer v_CPICH_Minus60, v_PCCPCH_Minus62,v_CPICH_Minus64, v_PCCPCH_Minus66;

    var UtraServingCellPriorityValue_Type v_UtraPriorityValue := 3;

    var EUTRA_MeasurementBandwidth v_EUTRA_MeasurementBandwidth := f_UTRAN_ConvertIntegerToEUTRA_MeasurementBandwidth(f_ConvertDL_BandwidthPIXITToInteger());

    var B28_Type v_CellIdentity := f_UTRAN_GetCellIndentity(utran_Cell5);

    var CSG_Identity v_CSG_Identity_Cell5 :='000000000000000000000000010'B;

    var octetstring v_PLMN := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran_Cell5));
    var InitialUE_Identity v_InitialUE_Id;

    var RLC_TR_DATA_IND v_RLC_TR_DATA_IND;

    timer t_WaitTimer := 360.0;

    v_CPICH_Minus60 := -60;

    v_PCCPCH_Minus62 := -62;

    v_CPICH_Minus64 := -64;

    v_PCCPCH_Minus66 := -66;

    // Initialise all cells

    f_UTRAN_Init(EUTRA_UTRAN);

    // Send the cell info to the EUTRA component

    f_UTRAN_Send_IR_SysInfoToEUTRA(utran_Cell5);

    // Get Cell Parameters for EUTRA cell

    v_EUTRASysInfo := f_IRAT_WaitForCoOrd_SysInfo(EUTRA);

    // Update SIB3 of cell 5 according to Table 6.4.3.3.3-3

    f_UTRAN_SysInfo_SetSIB3(utran_Cell5,

                            cds_SIB3_Csg(v_CellIdentity, v_CSG_Identity_Cell5, cs_CSG_PSCSplitInfo(13, psc5)));

    // Update SIB19 of cell 5 and cell 7 according to Table 6.4.3.3.3-4

    f_UTRAN_SysInfo_SetSIB19(utran_Cell5,

                             cs_SIB19_UtraAndOneEutraEntry(cs_Utra_PriorityInfoList_Def(v_UtraPriorityValue),

                                                           cs_Eutra_FreqAndPriList_OneEntry_Def(v_EUTRASysInfo.Eutra[0].Arfcn,

                                                                                                v_EUTRA_MeasurementBandwidth)));

    f_UTRAN_SysInfo_SetSIB19(utran_Cell7,

                             cs_SIB19_UtraAndOneEutraEntry(cs_Utra_PriorityInfoList_Def(v_UtraPriorityValue),

                                                           cs_Eutra_FreqAndPriList_OneEntry_Def(v_EUTRASysInfo.Eutra[0].Arfcn,

                                                                                                v_EUTRA_MeasurementBandwidth)));

    // Create and configure Cell 5 and cell 7

    f_UTRAN_SS_CreateCellFACH(utran_Cell5);

    f_UTRAN_SS_CreateCellFACH(utran_Cell7);

    // Start sending System Information

    f_UTRAN_SendDefSysInfo(utran_Cell5);

    f_UTRAN_SendDefSysInfo(utran_Cell7);

    if (pc_Manual_CSG_Selection)  {    //@sic R5-134565 sic@

      // Insert a USIM

      f_UT_USIM_Insert(UT, "USIM with service n86 available");

      // Empty UE Allowed CSG list in USIM

      //Update MIB of cell 5 to set it to CSG cell

      f_UTRAN_SysInfo_SetMIB(utran_Cell5, cds_MIB_Csg(cs_UTRAN_HPLMN_Def));

      f_UTRAN_Preamble_EmptyCSG_AllowedList(utran_Cell5);

      // Broadcast correct CSG id

      f_UTRAN_SysInfo_SetSIB3_CSG_Identity(utran_Cell5, v_CSG_Identity_Cell5);

      f_UTRAN_SysInfoModifySIB3(utran_Cell5, f_UTRAN_SysInfo_GetSIB3(utran_Cell5));

      // The UE's Allowed CSG list contains the CSG ID of Cell 5.

      f_UT_SwitchOnUE(UT, false);

      f_UT_ManualPLMN_CSG(UT ,v_PLMN, v_CSG_Identity_Cell5);

      f_UTRAN_IdleUpdated(utran_Cell5);

      f_UTRAN_SwitchPower_Off(utran_Cell5, U1_IDLE, NotSpecified);

      f_UTRAN_SetCellPower(utran_Cell5, tsc_AttenuationCellOFF);

      // Set cell 5 to Hybrid cell

      f_UTRAN_SysInfo_SetMIB(utran_Cell5, cs_MIB_Def(cs_UTRAN_HPLMN_Def));

      f_UTRAN_SysInfoModifyMIB(utran_Cell5);

    } else {

      // Insert a USIM

      f_UT_USIM_Insert(UT, "USIM with service n86 available, UE Allowed CSG list contains the CSG ID of Cell 5");

    }

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);
    f_UTRAN_TestBody_Set(true);

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // @siclog "Step 1" siclog@

    // The SS changes Cell 5 and Cell 7 levels according to the row "T1" in table 6.4.3.3.2-2.

    f_UTRAN_SetCellPower(utran_Cell5, v_CPICH_Minus60, v_PCCPCH_Minus62);

    f_UTRAN_SetCellPower(utran_Cell7, v_CPICH_Minus64, v_PCCPCH_Minus66);

    // @siclog "Step 2" siclog@

    // UE performs manual CSG ID selection and select Cell 5.

    f_UT_ManualPLMN_CSG(UT,v_PLMN, v_CSG_Identity_Cell5);

    // @siclog "Step 3-12" siclog@

    // Steps 1 to 10 of the generic test procedure described in TS 36.508 subclause 6.4.2.8 are performed on Cell 5.

    // NOTE: The UE performs a RAU procedure and the RRC connection is released.

    f_UTRAN_508CheckCampOnNewUtranCell(utran_Cell5);

    // @siclog "Step 13" siclog@

    // The SS changes Cell 1, Cell 2, Cell 5 and Cell 7 levels according to the row "T2" in table 6.4.3.3.2-1 and table 6.4.3.3.2-2.

    f_UTRAN_SetCellPower(utran_Cell5, tsc_AttenuationCellOFF);

    f_UTRAN_SetCellPower(utran_Cell7, tsc_AttenuationCellOFF);

    f_UTRAN_SendAuthCoOrdParameters(EUTRA);
    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    // @siclog "Step 22" siclog@

    // The SS changes Cell 1, Cell 2 and Cell 7 levels according to the row "T3" in table 6.4.3.3.2-1 and table 6.4.3.3.2-2.

    f_UTRAN_SetCellPower(utran_Cell7, v_CPICH_Minus64, v_PCCPCH_Minus66);

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // @siclog "Step 29" siclog@

    // The SS changes Cell 2 and Cell 5 levels according to the row "T4" in table 6.4.3.3.2-1 and table 6.4.3.3.2-2.

    f_UTRAN_SetCellPower(utran_Cell5, v_CPICH_Minus60, v_PCCPCH_Minus62);

    // @siclog "Step 30-37" siclog@

    // Check: Does the test result of steps 1 to 8 of generic test procedure in TS 36.508 subclause 6.4.2.8 indicate that the UE is camped on Cell 5 within 6 mins from "T4" to the time RRC CONNECTION REQUEST received by SS?

    // NOTE: The UE performs a RAU procedure

    t_WaitTimer.start;

    alt {

      [] U_TM.receive(car_RRC_ConnReq(utran_Cell5,

                                      tsc_RB0,

                                      cr_108_RRC_ConnReq_r8(registration))) -> value v_RLC_TR_DATA_IND
        {

          t_WaitTimer.stop;

          v_InitialUE_Id := v_RLC_TR_DATA_IND.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;

          f_UTRAN_RRC_ConnEst_WithoutRRCConnectionReq(utran_Cell5, v_InitialUE_Id);

          f_UTRAN_GMM_RAU(utran_Cell5);

          f_UTRAN_RRC_ConnRel(utran_Cell5, cell_Fach_Dcch);

          f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.4.3 Step 30-37");

        }

      [] t_WaitTimer.timeout{

        f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 6.4.3 Step 30-37");

      }

    }

    f_UTRAN_TestBody_Set(false);

    // Test case body finished, so let the other PTCs know

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger);

    // Release the UTRA cells

    // f_UTRAN_BringUSIMtoInital_CSG({utran_Cell5}); should use Detach and with cause "Not authorized for this CSG".

    f_UTRAN_Postamble(utran_Cell5, U1_IDLE);

    f_UTRAN_ReleaseCell(utran_Cell7, f_UTRAN_CellInfo_GetConfigType(utran_Cell7));

  } // end of f_TC_6_4_3_UTRAN

...

After change

...

  //========================================================================

  /*

   * @desc      REFERENCE TS 36.523-1 clause 6.4.3

   * @status

   */

  function f_TC_6_4_3_UTRAN() runs on UTRAN_PTC

  { // Inter-RAT cell reselection / From E-UTRA RRC_IDLE non-CSG cell to UTRA_Idle member hybrid cell

    var IRAT_CoOrd_SysInfo_Type v_EUTRASysInfo;

    var integer v_CPICH_Minus60, v_PCCPCH_Minus62,v_CPICH_Minus64, v_PCCPCH_Minus66;

    var UtraServingCellPriorityValue_Type v_UtraPriorityValue := 3;

    var EUTRA_MeasurementBandwidth v_EUTRA_MeasurementBandwidth := f_UTRAN_ConvertIntegerToEUTRA_MeasurementBandwidth(f_ConvertDL_BandwidthPIXITToInteger());

    var B28_Type v_CellIdentity := f_UTRAN_GetCellIndentity(utran_Cell5);

    var CSG_Identity v_CSG_Identity_Cell5 :='000000000000000000000000010'B;

    var octetstring v_PLMN;  
    var InitialUE_Identity v_InitialUE_Id;

    var RLC_TR_DATA_IND v_RLC_TR_DATA_IND;

    timer t_WaitTimer := 360.0;

    v_CPICH_Minus60 := -60;

    v_PCCPCH_Minus62 := -62;

    v_CPICH_Minus64 := -64;

    v_PCCPCH_Minus66 := -66;

    // Initialise all cells

    f_UTRAN_Init(EUTRA_UTRAN);

    v_PLMN := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran_Cell5));
    // Send the cell info to the EUTRA component

    f_UTRAN_Send_IR_SysInfoToEUTRA(utran_Cell5);

    // Get Cell Parameters for EUTRA cell

    v_EUTRASysInfo := f_IRAT_WaitForCoOrd_SysInfo(EUTRA);

    // Update SIB3 of cell 5 according to Table 6.4.3.3.3-3

    f_UTRAN_SysInfo_SetSIB3(utran_Cell5,

                            cds_SIB3_Csg(v_CellIdentity, v_CSG_Identity_Cell5, cs_CSG_PSCSplitInfo(13, psc5)));

    // Update SIB19 of cell 5 and cell 7 according to Table 6.4.3.3.3-4

    f_UTRAN_SysInfo_SetSIB19(utran_Cell5,

                             cs_SIB19_UtraAndOneEutraEntry(cs_Utra_PriorityInfoList_Def(v_UtraPriorityValue),

                                                           cs_Eutra_FreqAndPriList_OneEntry_Def(v_EUTRASysInfo.Eutra[0].Arfcn,

                                                                                                v_EUTRA_MeasurementBandwidth)));

    f_UTRAN_SysInfo_SetSIB19(utran_Cell7,

                             cs_SIB19_UtraAndOneEutraEntry(cs_Utra_PriorityInfoList_Def(v_UtraPriorityValue),

                                                           cs_Eutra_FreqAndPriList_OneEntry_Def(v_EUTRASysInfo.Eutra[0].Arfcn,

                                                                                                v_EUTRA_MeasurementBandwidth)));

    // Create and configure Cell 5 and cell 7

    f_UTRAN_SS_CreateCellFACH(utran_Cell5);

    f_UTRAN_SS_CreateCellFACH(utran_Cell7);

    // Start sending System Information

    f_UTRAN_SendDefSysInfo(utran_Cell5);

    f_UTRAN_SendDefSysInfo(utran_Cell7);

    if (pc_Manual_CSG_Selection)  {    //@sic R5-134565 sic@

      // Insert a USIM

      f_UT_USIM_Insert(UT, "USIM with service n86 available");

      // Empty UE Allowed CSG list in USIM

      //Update MIB of cell 5 to set it to CSG cell

      f_UTRAN_SysInfo_SetMIB(utran_Cell5, cds_MIB_Csg(cs_UTRAN_HPLMN_Def));

      f_UTRAN_Preamble_EmptyCSG_AllowedList(utran_Cell5);

      // Broadcast correct CSG id

      f_UTRAN_SysInfo_SetSIB3_CSG_Identity(utran_Cell5, v_CSG_Identity_Cell5);

      f_UTRAN_SysInfoModifySIB3(utran_Cell5, f_UTRAN_SysInfo_GetSIB3(utran_Cell5));

      // The UE's Allowed CSG list contains the CSG ID of Cell 5.

      f_UT_SwitchOnUE(UT, false);

      f_UT_ManualPLMN_CSG(UT ,v_PLMN, v_CSG_Identity_Cell5);

      f_UTRAN_IdleUpdated(utran_Cell5);

      f_UTRAN_SwitchPower_Off(utran_Cell5, U1_IDLE, NotSpecified);

      f_UTRAN_SetCellPower(utran_Cell5, tsc_AttenuationCellOFF);

      // Set cell 5 to Hybrid cell

      f_UTRAN_SysInfo_SetMIB(utran_Cell5, cs_MIB_Def(cs_UTRAN_HPLMN_Def));

      f_UTRAN_SysInfoModifyMIB(utran_Cell5);

    } else {

      // Insert a USIM

      f_UT_USIM_Insert(UT, "USIM with service n86 available, UE Allowed CSG list contains the CSG ID of Cell 5");

    }

    //f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); 
    //f_IRAT_WaitForCoOrd_Trigger(EUTRA); 

    // Coordination with LTE to start preamble in LTE

    f_UTRAN_SendAuthCoOrdParameters(EUTRA); 
    f_UTRAN_TestBody_Set(true);

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // @siclog "Step 1" siclog@

    // The SS changes Cell 5 and Cell 7 levels according to the row "T1" in table 6.4.3.3.2-2.

    f_UTRAN_SetCellPower(utran_Cell5, v_CPICH_Minus60, v_PCCPCH_Minus62);

    f_UTRAN_SetCellPower(utran_Cell7, v_CPICH_Minus64, v_PCCPCH_Minus66);

    // @siclog "Step 2" siclog@

    // UE performs manual CSG ID selection and select Cell 5.

    f_UT_ManualPLMN_CSG(UT,v_PLMN, v_CSG_Identity_Cell5);

    // @siclog "Step 3-12" siclog@

    // Steps 1 to 10 of the generic test procedure described in TS 36.508 subclause 6.4.2.8 are performed on Cell 5.

    // NOTE: The UE performs a RAU procedure and the RRC connection is released.

    f_UTRAN_508CheckCampOnNewUtranCell(utran_Cell5);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA); 

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); 
    // @siclog "Step 13" siclog@

    // The SS changes Cell 1, Cell 2, Cell 5 and Cell 7 levels according to the row "T2" in table 6.4.3.3.2-1 and table 6.4.3.3.2-2.

    f_UTRAN_SetCellPower(utran_Cell5, tsc_AttenuationCellOFF);

    f_UTRAN_SetCellPower(utran_Cell7, tsc_AttenuationCellOFF);

    //f_UTRAN_SendAuthCoOrdParameters(EUTRA); 
    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    // @siclog "Step 22" siclog@

    // The SS changes Cell 1, Cell 2 and Cell 7 levels according to the row "T3" in table 6.4.3.3.2-1 and table 6.4.3.3.2-2.

    f_UTRAN_SetCellPower(utran_Cell7, v_CPICH_Minus64, v_PCCPCH_Minus66);

    f_UTRAN_InterRAT_InitialiseAuthParams();

    // @siclog "Step 29" siclog@

    // The SS changes Cell 2 and Cell 5 levels according to the row "T4" in table 6.4.3.3.2-1 and table 6.4.3.3.2-2.

    f_UTRAN_SetCellPower(utran_Cell5, v_CPICH_Minus60, v_PCCPCH_Minus62);

    // @siclog "Step 30-37" siclog@

    // Check: Does the test result of steps 1 to 8 of generic test procedure in TS 36.508 subclause 6.4.2.8 indicate that the UE is camped on Cell 5 within 6 mins from "T4" to the time RRC CONNECTION REQUEST received by SS?

    // NOTE: The UE performs a RAU procedure

    t_WaitTimer.start;

    alt {

      [] U_TM.receive(car_RRC_ConnReq(utran_Cell5,

                                      tsc_RB0,

                                      cr_108_RRC_ConnReq_r8(cr_EstCause_Reselection))) -> value v_RLC_TR_DATA_IND  
        {

          t_WaitTimer.stop;

          v_InitialUE_Id := v_RLC_TR_DATA_IND.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;

          f_UTRAN_RRC_ConnEst_WithoutRRCConnectionReq(utran_Cell5, v_InitialUE_Id);

          f_UTRAN_GMM_RAU(utran_Cell5);

          f_UTRAN_RRC_ConnRel(utran_Cell5, cell_Fach_Dcch);

          f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.4.3 Step 30-37");

        }

      [] t_WaitTimer.timeout{

        f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 6.4.3 Step 30-37");

      }

    }

    f_UTRAN_TestBody_Set(false);

    // Test case body finished, so let the other PTCs know

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger);

    // Release the UTRA cells

    // f_UTRAN_BringUSIMtoInital_CSG({utran_Cell5}); should use Detach and with cause "Not authorized for this CSG".

    f_UTRAN_Postamble(utran_Cell5, U1_IDLE);

    f_UTRAN_ReleaseCell(utran_Cell7, f_UTRAN_CellInfo_GetConfigType(utran_Cell7));

  } // end of f_TC_6_4_3_UTRAN

...

Change 3 – Correction to UTRAN CSG functions
See change 3 in R5s140372.

Change 4 – Introduction of new UTRAN RRC template
See change 4 in R5s140372.

Change 5 – Correction to ‘t_GuardTimer’ for TC 6.4.3
	TC name
	testcase TC_6_4_3()

	Reason for change
	Guard timer of 360 sec. is too short to cater for the manual CSG ID selection and the SI reading period each time the cell power level are changed.

	Summary of change
	Increased guard timer to 700 sec.

	TTCN module
	LTE_IRAT_Testsuite.ttcn

	MCC160 Comment
	


Before change

...

  //----------------------------------------------------------------------------

  //----------------------------------------------------------------------------
  testcase TC_6_4_3() runs on MTC_LTE_IRAT system SYSTEM_LTE_IRAT {

    // @purpose

    //   Inter-RAT cell reselection / From E-UTRA RRC_IDLE non-CSG cell to UTRA_Idle member hybrid cell

    var EUTRA_PTC    v_EUTRA      := null;

    var UTRAN_PTC    v_UTRAN      := null;

    var GERAN_PTC    v_GERAN      := null;

    var CDMA2000_PTC v_CDMA2000   := null;

    var IMS_PTC      v_IMS1       := null;

    var IMS_PTC      v_IMS2       := null;

    timer t_GuardTimer := int2float(360); 

    v_EUTRA := EUTRA_PTC.create alive;

    v_UTRAN := UTRAN_PTC.create alive;

    f_MTC_ConnectPTCs_LTE_IRAT(v_EUTRA, v_UTRAN, v_GERAN, v_CDMA2000, v_IMS1, v_IMS2);

    v_EUTRA.start(f_TC_6_4_3_EUTRA());

    v_UTRAN.start(f_TC_6_4_3_UTRAN());

    t_GuardTimer.start;

    f_MTC_MainLoop(t_GuardTimer);

  }

  //----------------------------------------------------------------------------

  //----------------------------------------------------------------------------

... 
After change

...

  //----------------------------------------------------------------------------

  //----------------------------------------------------------------------------
  testcase TC_6_4_3() runs on MTC_LTE_IRAT system SYSTEM_LTE_IRAT {

    // @purpose

    //   Inter-RAT cell reselection / From E-UTRA RRC_IDLE non-CSG cell to UTRA_Idle member hybrid cell

    var EUTRA_PTC    v_EUTRA      := null;

    var UTRAN_PTC    v_UTRAN      := null;

    var GERAN_PTC    v_GERAN      := null;

    var CDMA2000_PTC v_CDMA2000   := null;

    var IMS_PTC      v_IMS1       := null;

    var IMS_PTC      v_IMS2       := null;

    timer t_GuardTimer := int2float(700);  

    v_EUTRA := EUTRA_PTC.create alive;

    v_UTRAN := UTRAN_PTC.create alive;

    f_MTC_ConnectPTCs_LTE_IRAT(v_EUTRA, v_UTRAN, v_GERAN, v_CDMA2000, v_IMS1, v_IMS2);

    v_EUTRA.start(f_TC_6_4_3_EUTRA());

    v_UTRAN.start(f_TC_6_4_3_UTRAN());

    t_GuardTimer.start;

    f_MTC_MainLoop(t_GuardTimer);

  }

  //----------------------------------------------------------------------------

  //----------------------------------------------------------------------------

... 
5. Execution Log Files

5.1 Qualcomm MSM 8974
The Qualcomm MSM 8974 UE passed this test case on R&S CMW500 in LTE FDD band 4 and UTRAN band II. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test Case Execution log file: 
Rohde-Schwarz\tc_6_4_3_Qualcomm.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)


In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

5.2 Qualcomm 9X25
The Qualcomm 9X25 UE passed this test case on Anite Platform in LTE FDD band 4 and UTRAN band II. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test Case Execution log file: 
Anite\Anite_tc_6_4_3_Qualcomm.log   
(Note: PICS/PIXIT settings are captured in separate file “6_4_3_PICS_PIXIT.xml”)

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

References

	R5s140375:    Supporting information for agreement of TC 6.4.3 in FDD mode.

