
3. Corrections

Change 1 – Corrections to function ‘f_IMS_SRVCC_ReInvite_AnnexC31_Step1’
	Function name
	f_IMS_SRVCC_ReInvite_AnnexC31_Step1()

	Reason for change
	1.    Annex C.31 was introduced at RAN5 #62 with R5-140799, but it was not considered that only as of Rel-10 of TS 24.237 section 12.2.4 in 1), it is for abnormal cases mandatory to specify a reason-text in the “INVITE for MO Call”. The test case implementation may therefore unfairly FAIL a Rel-9 or earlier UE. 
(see attached draft Prose CR “draft_R5-142xxx_34229-1_Annex-C31”)
2.    According to TS 34.229 A.2.5 condition A5 should apply: “value of CSeq sent by the UE within its previous request in the same dialog but increased by one”. In this scenario, Re-Invite message is the first request coming from the UE, therefore it should not be checked against any previous value stored in the IMS PTC.

	Summary of change
	1.    Made checking for reason-text dependent on UE Release.
2.    Added new boolean for distinguis Re-Invite message when the call is MT and the Re-Invite MO. 

	TTCN module
	IMS_Procedures_CallControl.ttcn

	MCC160 Comment
	


Before:

	…
function f_IMS_SRVCC_ReInvite_AnnexC31_Step1(boolean p_GeolocationInfoAvailable := false)

    runs on IMS_PTC return IMS_DATA_REQ

  { /* @sic R5s140020, R5-140799: new function sic@ */

    /* @sic R5-140799 sic@ */

    var template (omit) float v_WaitDuration := omit;   // no wait duration

    var template (omit) integer v_UE_Release := f_IMS_PTC_GetUE_Release();

    var template (present) SDP_Message v_SDP_MessageRX := f_IMS_BuildSDP_AnnexC31_Step1();

    var template (present) MessageBody v_MessageBody := f_SIP_MessageBodySDP_RX(v_SDP_MessageRX, p_GeolocationInfoAvailable);

    var template (present) MessageHeader v_MessageHeader := cr_MessageHeader_Dummy;

    var template charstring v_ReasonText := ("failure to transition to CS domain", "handover cancelled");

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var SDP_Message v_SDP_Message;

/*     if (not (ispresent(v_UE_Release) and (valueof(v_UE_Release) >= 11))) { */

/*       v_ReasonText := omit; */

/*     } */
    v_MessageHeader.reason := cr_Reason(487, v_ReasonText);     // @sic R5-140799 sic@

    v_IMS_DATA_REQ := valueof(f_IMS_INVITE_ReceiveRequest_Common(A_2_1_A5, v_MessageHeader, -, v_MessageBody, -, v_WaitDuration, -, p_GeolocationInfoAvailable));

    v_SDP_Message := f_IMS_MessageBody_CheckGeolocationAndGetSDP(v_IMS_DATA_REQ.Request.Invite.messageBody, p_GeolocationInfoAvailable);

    // check at least one connections field sent

    f_IMS_SDP_CheckConnections(v_SDP_Message);

    return v_IMS_DATA_REQ;

  }   
…


After:

	…
function f_IMS_SRVCC_ReInvite_AnnexC31_Step1(boolean p_GeolocationInfoAvailable := false)

    runs on IMS_PTC return IMS_DATA_REQ

  { /* @sic R5s140020, R5-140799: new function sic@ */

    /* @sic R5-140799 sic@ */

    var template (omit) float v_WaitDuration := omit;   // no wait duration

    var template (omit) integer v_UE_Release := f_IMS_PTC_GetUE_Release();

    var template (present) SDP_Message v_SDP_MessageRX := f_IMS_BuildSDP_AnnexC31_Step1();

    var template (present) MessageBody v_MessageBody := f_SIP_MessageBodySDP_RX(v_SDP_MessageRX, p_GeolocationInfoAvailable);

    var template (present) MessageHeader v_MessageHeader := cr_MessageHeader_Dummy;

    var template charstring v_ReasonText := ("failure to transition to CS domain", "handover cancelled");

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var SDP_Message v_SDP_Message;

    //Review R5-140799

    if (not (ispresent(v_UE_Release) and (valueof(v_UE_Release) >= 10))) { 

       v_ReasonText := omit; 

     } 

    v_MessageHeader.reason := cr_Reason(487, v_ReasonText);     // @sic R5-140799 sic@

    v_IMS_DATA_REQ := valueof(f_IMS_INVITE_ReceiveRequest_Common(A_2_1_A5, v_MessageHeader, -, v_MessageBody, -, v_WaitDuration, -, p_GeolocationInfoAvailable,-,-,-,-,true));
    v_SDP_Message := f_IMS_MessageBody_CheckGeolocationAndGetSDP(v_IMS_DATA_REQ.Request.Invite.messageBody, p_GeolocationInfoAvailable);

    // check at least one connections field sent

    f_IMS_SDP_CheckConnections(v_SDP_Message);

    return v_IMS_DATA_REQ;

  }  }    

…


Change 2 – Corrections to function ‘f_IMS_InviteRequest_MessageHeaderRX’
	Function name
	f_IMS_InviteRequest_MessageHeaderRX()

	Reason for change
	See change 1.2 above.

	Summary of change
	1.
New boolean added.

2.
Received CSeq number value modified to any value. 

	TTCN module
	IMS_Procedures_CallControl.ttcn

	MCC160 Comment
	


Before:

	…

  function f_IMS_InviteRequest_MessageHeaderRX(template (present) MessageHeader p_MessageHeader := cr_MessageHeader_Dummy,

                                               template (present) SipUrl p_ExpectedCalleeUri,

                                               OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                               INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                               boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                               boolean p_A8 := pc_IMS_Geolocation,

                                               boolean p_A9 := false,

                                               boolean p_A11 := false)

    runs on IMS_PTC return template (present) MessageHeader

  { /* p_A3:          MTSI

       p_A8:          UE uses Geolocation header to provide its geographical location for emergency session setup, has obtained its location and is setting up an emergency session

       p_A9:          UE supports Session-ID

       p_A11:         INVITE for creating a video call and UE supports video media feature tag

       NOTE1: If SigComp shall be applied the param "comp=sigcomp" is optional for Via, Route and Contact

       NOTE2: condition A5 (re-INVITE within a dialog) is not considered yet and will have impact e.g. on checking of the From/To headers */

    /* @sic R5-133629: 100rel option tag shall always be included in INVITE; further option tags may be added => p_Supported replaced by p_AdditionalOptionTagsForSupported; Cleanup of functions for MO Call Setup sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    /* @sic R5s140017 MCC160 implementation: checks for A_2_1_A5 sic@ */

    /* @sic R5-140799: additional parameter p_MessageHeader to deal with Reason header sic@ */

    var boolean v_IsGiba := f_IMS_PTC_SecurityScheme_IsGiba();  // => A1/A2

    var template (present) MessageHeader v_MessageHeader_Invite := p_MessageHeader;

    var template (present) RouteBody_List v_RouteBodyList;

    var IMS_ProtectedPorts_Type v_Protected;

    var template integer v_Port_us := *;

    var template integer v_Port_ps := *;

    var template charstring v_Displayname := *;

    var template (present) SipUrl v_SipUrlFrom := ?;    /* per default any SIP URI is accepted in the From header in the first place;

                                                           further checks may be applied in f_IMS_A_2_1_Invite_CommonChecks (depending on the conditions acc. to A.2.1 */

    var template (present) charstring v_MediaType := "application/sdp";

    var template (present) CallId v_CallId;

    var template (present) integer v_CSeqValue;

    var template (present) From v_From;

    var template (present) To v_To;

    var template SessionId v_SessionId;

    var OptionTag_List v_OptionTag_List := tsc_OptionTagList_100rel & p_AdditionalOptionTagsForSupported;

    if (p_Context == A_2_1_A6) {  // INVITE for creating an emergency session in case of no registration

      // Via, Route, Contact: ports to be used

      v_Port_us := ?;

      v_Port_ps := tsc_IMS_PortNumber_5060;

      // From: Display

      v_Displayname := cr_QuotedStringOrToken("Anonymous");     /* @sic R5s140005, R5-140901 sic@ */

    }

    else {

      // Via, Route, Contact: ports to be used

      if (not v_IsGiba or (p_Context == A_2_1_A7)) {

        v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

        v_Port_us := v_Protected.Port_us;

        v_Port_ps := v_Protected.Port_ps;

      }

    }

    if (p_Context == A_2_1_A5)  {                  /* re-invite @sic R5s140017 MCC160 implementation sic@ */

      v_RouteBodyList := fl_RouteSet_RouteBodyList_RX();

      v_CallId := f_IMS_PTC_ImsInfo_DialogGetCallId();

      v_From := f_IMS_RequestInDialog_FromHeaderRX();

      v_To := f_IMS_RequestInDialog_ToHeaderRX();

      v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogRemote);
      v_SessionId := f_IMS_PTC_ImsInfo_DialogGetSessionId();

    }

    else {                                       /* normal case: no re-invite */

      // route body acc. to conditions in A.2.1

      v_RouteBodyList[0] := cr_RouteBody(*, cr_SipUri_HostPort_lr(f_IMS_PTC_Pcscf_Get(), v_Port_ps));

      if (not match(p_Context, (A_2_1_A6, A_2_1_A7))) {  // => no emergency call

        v_RouteBodyList[1] := cr_RouteBody(*, cr_SipUri_HostPort_lr(px_IMS_Scscf));

      }

      if (p_A8) {

        v_MessageHeader_Invite.geolocation := cr_GeolocationAny;

        v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingSupported; // @sic R5-134798 sic@

        v_MediaType := "multipart/mixed";

      }

      v_CallId := cr_CallId(?);

      v_CSeqValue := ?;

      v_From := cr_From(v_Displayname, v_SipUrlFrom);

      v_To := cr_ToDef(p_ExpectedCalleeUri);

      v_SessionId := f_IMS_SessionId_RX_IfSupported(p_A9);

    }

    v_MessageHeader_Invite.via             := fl_IMS_InviteRequest_ViaRX();                           /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address

                                                                                                         @sic R5-133358: rport is optional for tcp sic@ */

    v_MessageHeader_Invite.route           := cr_Route(v_RouteBodyList);

    v_MessageHeader_Invite.fromField       := v_From;                                                 /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address, check against P-Preferred-Identity header */

    v_MessageHeader_Invite.toField         := v_To;                                                   /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of tag */

    v_MessageHeader_Invite.callId          := v_CallId;                                               /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: value acc. to A4 or A5 */

    v_MessageHeader_Invite.sessionId       := v_SessionId;                                            /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Invite.cSeq            := cr_CseqDef(v_CSeqValue, "INVITE");                      /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: A5 */

    v_MessageHeader_Invite.supported       := cr_Supported(v_OptionTag_List);

    v_MessageHeader_Invite.contact         := cr_Contact(cr_SipUri_HostPort(?, v_Port_us));           /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address, check of feature-param for A3, A6, A10, A12 */

    v_MessageHeader_Invite.contentType     := cr_ContentType(v_MediaType);

    v_MessageHeader_Invite.maxForwards     := cr_MaxForwardsDef;

    // additional headers for specific conditions:

    if (v_IsGiba or (p_Context == A_2_1_A6)) {

      v_MessageHeader_Invite.securityVerify  := omit;

      if (p_Context == A_2_1_A6) {

        v_MessageHeader_Invite.require       := omit;

        v_MessageHeader_Invite.proxyRequire  := omit;

      }

    } else {   // -> A1 or A7

      v_MessageHeader_Invite.securityVerify  := f_IMS_PTC_BuildSecurityVerifyHeader();

      if (p_Context != A_2_1_A5) {  /* @sic R5s140017 change 9.3 sic@ */

        v_MessageHeader_Invite.require       := cr_RequireSecAgree;

        v_MessageHeader_Invite.proxyRequire  := cr_ProxyRequireDef;

      }

    }

    if (not v_IsGiba) {

      v_MessageHeader_Invite.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *);

    }

    if (p_Context == A_2_1_A4) {  /* @sic R5s140017 change 9.3 sic@ */

      v_MessageHeader_Invite.accept          := cr_Accept({cr_AcceptBody("application/sdp"), cr_AcceptBody("application/3gpp-ims+xml"), *});

    }

    if (p_A8) {

      v_MessageHeader_Invite.geolocation     := cr_GeolocationAny;  // Checked in f_IMS_A_2_1_Invite_CommonChecks

      v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingAny;  // Checked in f_IMS_A_2_1_Invite_CommonChecks @sic R5-133151 sic@

    }

    if (p_Context == A_2_1_A7) {

      v_MessageHeader_Invite.pPreferredID := cr_PPreferredID(?);    // Checked in f_IMS_A_2_1_Invite_CommonChecks

    }

    if (p_A3) {

      v_MessageHeader_Invite.pPreferredService := cr_PPreferredService(tsc_IMS_AcceptContactValue);

    }

    if (p_A3 or p_A11) {

      v_MessageHeader_Invite.acceptContact   := cr_AcceptContact;    // Checked in f_IMS_A_2_1_Invite_CommonChecks (depending on conditions)

    }

    return v_MessageHeader_Invite;

  }   
…


After:

	…
function f_IMS_InviteRequest_MessageHeaderRX(template (present) MessageHeader p_MessageHeader := cr_MessageHeader_Dummy,

                                               template (present) SipUrl p_ExpectedCalleeUri,

                                               OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                               INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                               boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                               boolean p_A8 := pc_IMS_Geolocation,

                                               boolean p_A9 := false,

                                               boolean p_A11 := false,

                                               boolean p_ReInviteMOMTcall := false) 

    runs on IMS_PTC return template (present) MessageHeader

  { /
….

    if (p_Context == A_2_1_A5)  {                  /* re-invite @sic R5s140017 MCC160 implementation sic@ */

      v_RouteBodyList := fl_RouteSet_RouteBodyList_RX();

      v_CallId := f_IMS_PTC_ImsInfo_DialogGetCallId();

      v_From := f_IMS_RequestInDialog_FromHeaderRX();

      v_To := f_IMS_RequestInDialog_ToHeaderRX();

      v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogRemote);

      v_SessionId := f_IMS_PTC_ImsInfo_DialogGetSessionId();

      if(p_ReInviteMOMTcall)

      {

      
 v_CSeqValue := ?; 

      }
    }

    else {                                       /* normal case: no re-invite */

      // route body acc. to conditions in A.2.1

      v_RouteBodyList[0] := cr_RouteBody(*, cr_SipUri_HostPort_lr(f_IMS_PTC_Pcscf_Get(), v_Port_ps));

      if (not match(p_Context, (A_2_1_A6, A_2_1_A7))) {  // => no emergency call

        v_RouteBodyList[1] := cr_RouteBody(*, cr_SipUri_HostPort_lr(px_IMS_Scscf));

      }

….

    return v_MessageHeader_Invite;

  }    

…


Change 3 – Corrections to function ‘f_IMS_INVITE_ReceiveRequest_Common’
	Function name
	f_IMS_INVITE_ReceiveRequest_Common()

	Reason for change
	See change 1.2 above.

	Summary of change
	1. New boolean added.
2. New parameter as input parameter.

	TTCN module
	IMS_Procedures_CallControl.ttcn

	MCC160 Comment
	


Before:

	…

  function f_IMS_INVITE_ReceiveRequest_Common(INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                              template (present) MessageHeader p_MessageHeader := cr_MessageHeader_Dummy,

                                              OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                              template (present) MessageBody p_MessageBody,

                                              template (omit) charstring p_OtherCalleeUri := omit,

                                              template (omit) float p_WaitDuration := omit,

                                              boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                              boolean p_A8 := pc_IMS_Geolocation,

                                              boolean p_A9 := false,

                                              boolean p_A10 := false,

                                              boolean p_A11 := false,

                                              boolean p_A12 := false)

    runs on IMS_PTC return template (omit) IMS_DATA_REQ

  { /* p_OtherCalleeUri .. if present to be used as CalleeUri; if omit default CalleeUri is used acc. to PIXITs and cntext */

    /* @sic R5-133276, R5-133301 Cleanup of functions for MO Call Setup sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    /* @sic R5-140799: additional parameter p_MessageHeader to deal with Reason header sic@ */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (present) MessageHeader v_MessageHeader_Invite;

    var template (present) SipUrl v_ExpectedCalleeUri;

    var boolean v_TimerControl := ispresent(p_WaitDuration);

    timer t_Timer;

    if (ispresent(p_OtherCalleeUri)) {

      v_ExpectedCalleeUri := f_SIP_BuildSipUri_RX(valueof(p_OtherCalleeUri));

    } else {

      select (p_Context) {

        case (A_2_1_A4) {                // A4: INVITE creating a dialog

          v_ExpectedCalleeUri := f_SIP_BuildSipUri_RX(px_IMS_CalleeUri);

        }

        case (A_2_1_A5) {                // A5: re-INVITE

          v_ExpectedCalleeUri := f_SIP_BuildSipUri_RX(px_IMS_CalleeContactUri, *);  /* @sic R5s140020 change 8.1 sic@ */

        }

        case (A_2_1_A6, A_2_1_A7) {      // A6, A7: emergency call

          v_ExpectedCalleeUri := cr_SipUrl_Common(c_urnScheme, cr_SipUrl_SOS);

        }

        case else {

          FatalError(__FILE__, __LINE__, "");

        }

      }

    }

    v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderRX(p_MessageHeader, v_ExpectedCalleeUri, p_AdditionalOptionTagsForSupported, p_Context, p_A3, p_A8, p_A9, p_A11);
    if (v_TimerControl) {

      t_Timer.start(valueof(p_WaitDuration));

    }

    alt {

      [] IMS_Server.receive(car_IMS_Invite_Request(cr_INVITE_Request(v_ExpectedCalleeUri, v_MessageHeader_Invite, p_MessageBody))) -> value v_IMS_DATA_REQ

        {

          t_Timer.stop;

          if (p_Context == A_2_1_A6) {      // emergency call without registration   @sic R5s130756 change 14 sic@

            f_IMS_PTC_ImsInfo_Init(v_IMS_DATA_REQ.RoutingInfo.UE_Address, v_IMS_DATA_REQ.RoutingInfo.NW_Address);

          }

          f_IMS_A_2_1_Invite_CommonChecks(v_IMS_DATA_REQ, p_Context, p_A3, p_A8, p_A10, p_A11, p_A12);

          if (p_Context != A_2_1_A5) {      /* @sic R5s140020 change 8.2 sic@ */

            f_IMS_PTC_ImsInfo_DialogInit(dialogRemote, v_IMS_DATA_REQ.Request.Invite.msgHeader);

          }

          return v_IMS_DATA_REQ;

        }

      [v_TimerControl] t_Timer.timeout {}

    }

    return omit;

  }   

…


After:

	…
function f_IMS_INVITE_ReceiveRequest_Common(INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                              template (present) MessageHeader p_MessageHeader := cr_MessageHeader_Dummy,

                                              OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                              template (present) MessageBody p_MessageBody,

                                              template (omit) charstring p_OtherCalleeUri := omit,

                                              template (omit) float p_WaitDuration := omit,

                                              boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                              boolean p_A8 := pc_IMS_Geolocation,

                                              boolean p_A9 := false,

                                              boolean p_A10 := false,

                                              boolean p_A11 := false,

                                              boolean p_A12 := false,

                                              boolean p_ReInviteMOMTcall := false) 

    runs on IMS_PTC return template (omit) IMS_DATA_REQ

  { /* p_OtherCalleeUri .. if present to be used as CalleeUri; if omit default CalleeUri is used acc. to PIXITs and cntext */

    /* @sic R5-133276, R5-133301 Cleanup of functions for MO Call Setup sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    /* @sic R5-140799: additional parameter p_MessageHeader to deal with Reason header sic@ */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (present) MessageHeader v_MessageHeader_Invite;

    var template (present) SipUrl v_ExpectedCalleeUri;

    var boolean v_TimerControl := ispresent(p_WaitDuration);

    timer t_Timer;

    if (ispresent(p_OtherCalleeUri)) {

      v_ExpectedCalleeUri := f_SIP_BuildSipUri_RX(valueof(p_OtherCalleeUri));

    } else {

      select (p_Context) {

        case (A_2_1_A4) {                // A4: INVITE creating a dialog

          v_ExpectedCalleeUri := f_SIP_BuildSipUri_RX(px_IMS_CalleeUri);

        }

        case (A_2_1_A5) {                // A5: re-INVITE

          v_ExpectedCalleeUri := f_SIP_BuildSipUri_RX(px_IMS_CalleeContactUri, *);  /* @sic R5s140020 change 8.1 sic@ */

        }

        case (A_2_1_A6, A_2_1_A7) {      // A6, A7: emergency call

          v_ExpectedCalleeUri := cr_SipUrl_Common(c_urnScheme, cr_SipUrl_SOS);

        }

        case else {

          FatalError(__FILE__, __LINE__, "");

        }

      }

    }

    v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderRX(p_MessageHeader, v_ExpectedCalleeUri, p_AdditionalOptionTagsForSupported, p_Context, p_A3, p_A8, p_A9, p_A11, p_ReInviteMOMTcall); 

    if (v_TimerControl) {

      t_Timer.start(valueof(p_WaitDuration));

    }

    alt {

      [] IMS_Server.receive(car_IMS_Invite_Request(cr_INVITE_Request(v_ExpectedCalleeUri, v_MessageHeader_Invite, p_MessageBody))) -> value v_IMS_DATA_REQ

        {

          t_Timer.stop;

          if (p_Context == A_2_1_A6) {      // emergency call without registration   @sic R5s130756 change 14 sic@

            f_IMS_PTC_ImsInfo_Init(v_IMS_DATA_REQ.RoutingInfo.UE_Address, v_IMS_DATA_REQ.RoutingInfo.NW_Address);

          }

          f_IMS_A_2_1_Invite_CommonChecks(v_IMS_DATA_REQ, p_Context, p_A3, p_A8, p_A10, p_A11, p_A12);

          if (p_Context != A_2_1_A5) {      /* @sic R5s140020 change 8.2 sic@ */

            f_IMS_PTC_ImsInfo_DialogInit(dialogRemote, v_IMS_DATA_REQ.Request.Invite.msgHeader);

          }

          return v_IMS_DATA_REQ;

        }

      [v_TimerControl] t_Timer.timeout {}

    }

    return omit;

  }    
…


