
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 13.4.3.6 which is part of the LTE test suite ‘iwd-EUTRA-B2013-03_D13k49’.

The test case can be demonstrated to run with one LTE FDD UE (see section 5). Execution logs are provided as evidence.

3. Verification Test Summary

Test Case:
TC_13_4_3_6
Test Group:
LTE_IRAT_IWD_13wk49
ATS Version:
iwd-EUTRA-B2013-03_D13wk49
System Simulator used:
Anite Conformance Toolset solution
UE used:
Qualcomm MSM 9x25
Verification Status:
PASS


4. Corrections required for test case 13.4.3.6
This section describes the TTCN change required to make test case 13.4.3.6 run correctly with an LTE UE. The TTCN used to pass this test case belongs to ‘iwd-EUTRA-B2013-03_D13wk49’ release.

4.1 Change 1
	Testcase name
	13.4.3.6

	Protocol
	Multilayer

	Function Name
	f_TC_13_4_3_6_EUTRA()

	Reason for change
	1. Incorrect cell ID used in call to f_EUTRA_CellInfo_InitMaxReferencePower()

2. Power level of EUTRA cell is not set to T0 level following preamble

3. On completion of the test case body, the UE will be in CONNECTED state on the EUTRA cell, howver the current postamble handling releases the EUTRA cell without any signalling to the UE. Also, we should send a trigger to UTRAN PTC to mark the end of the test case rather than waiting for one.

	Summary of change
	1. Correct cell ID used in call to f_EUTRA_CellInfo_InitMaxReferencePower()

2. Add call to f_EUTRA_SetCellPower() following preamble

3. Add correct postamble function call and corrected coordination triggers.

	Source of change
	SRVCC_EUTRA.ttcn


Before:
	    function f_TC_13_4_3_6_EUTRA() runs on EUTRA_PTC

  { /* Inter-system mobility / E-UTRA PS voice + PS Data / HO cancelled / Notification procedure / SRVCC */

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c4);

    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell2, -60); //@sic R5-125740 sic@

    // Create cell and send out system information

    f_EUTRA_CellConfig_DefPlusUM_RoHC(eutra_Cell1);

    //f_IRAT_WaitForCoOrd_Trigger(UTRAN); // wait for UTRAN side to do pre-preamble @sic R5s110776 sic@

    f_EUTRA_Preamble_ASRVCC (eutra_Cell1, STATE2_IDLEUPDATE);

    f_EUTRA_TestBody_Set(true);

    fl_TC_13_4_3_6_Body();

     f_IRAT_WaitForCoOrd_Trigger(UTRAN);
    f_EUTRA_TestBody_Set(false);

    f_EUTRA_ReleaseAllCells ();
  } // function f_TC_13_4_3_6_EUTRA  


After:
	    function f_TC_13_4_3_6_EUTRA() runs on EUTRA_PTC

  { /* Inter-system mobility / E-UTRA PS voice + PS Data / HO cancelled / Notification procedure / SRVCC */

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c4);

    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell1, -60); //@sic R5-125740 sic@

    // Create cell and send out system information

    f_EUTRA_CellConfig_DefPlusUM_RoHC(eutra_Cell1);

    //f_IRAT_WaitForCoOrd_Trigger(UTRAN); // wait for UTRAN side to do pre-preamble @sic R5s110776 sic@

    f_EUTRA_Preamble_ASRVCC (eutra_Cell1, STATE2_IDLEUPDATE);

    f_EUTRA_SetCellPower(eutra_Cell1, -60);
    f_EUTRA_TestBody_Set(true);

    fl_TC_13_4_3_6_Body();

    // Postamble

    f_EUTRA_TestBody_Set(false);

    f_IRAT_SendCoOrd(UTRAN, cms_IRAT_Trigger); //

    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]);// Send co-ordination message to receive any De Register message

    // Switch off and release cells

    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);
  } // function f_TC_13_4_3_6_EUTRA



4.2 Change 2
	Testcase name
	13.4.3.6

	Protocol
	Multilayer

	Function Name
	fl_TC_13_4_3_6_Body()

	Reason for change
	1. When sending the coordination trigger to UTRA PTC to set T0 power levels it is not required to wait for a response
2. Measurement control configured is configuring RSCP reporting values for UTRAN, but the receive template for measurement report is expecting EcNo values.
3. Incorrect cell ID used for sending NOTIFICATION message

4. Incorrect transaction ID used when sending NOTIFICATION message
5. SRB / DRB release not needed here as full postamble sequence is performed in the calling function (see change 1)

	Summary of change
	1. Use “false” in call to f_EUTRA_SendTriggerToOtherRAT()
2. Use template cr_MeasurementReport_Utran_1Entry_FDD_RSCP to receive measurement report
3. Use cell ID 1 to send NOTIFICATION message
4. Use transaction ID 0 to send NOTIFICATION message

5. Remove call to  f_EUTRA_SS_SRBs_DRBs_Release()

	Source of change
	SRVCC_EUTRA.ttcn


Before:
	      function fl_TC_13_4_3_6_Body() runs on EUTRA_PTC

  {

    var AbsoluteCellPower_Type v_T1_RS_EPRE_Cell1 := -84;

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var integer v_UTRAN_PhysChId_Cell5;

    var octetstring v_HandoverToUtranCommand;

    //var integer v_UTRAN_ARFCN_f8; //Cell5

    var Frequency_IE_Type   v_Frequency_IE_Cell1 := f_EUTRA_CellInfo_GetFrequencyIEs (eutra_Cell1);

    var ChannelBandwidthDependency_Type  v_ChBandDependency_Cell1:= f_EUTRA_BandDependentParam(v_Frequency_IE_Cell1.DL_ChBandwidth, v_Frequency_IE_Cell1.UL_ChBandwidth);

    var integer v_UTRAN_ARFCN_Cell5;

    var IRAT_CoOrd_SysInfo_Type v_UTRANSysInfo;

    var CoOrd_UTRANSysInfoFDD_Type v_UTRANSysInfoFDD;

    var CoOrd_UTRANSysInfoTDD_Type v_UTRANSysInfoTDD; 

    v_UTRANSysInfo := f_IRAT_WaitForCoOrd_SysInfo (UTRAN);

    if (ischosen (v_UTRANSysInfo.Utran[0].FDD)) {

      v_UTRANSysInfoFDD := v_UTRANSysInfo.Utran[0].FDD;

        v_UTRAN_ARFCN_Cell5 := v_UTRANSysInfo.Utran[0].FDD.Freq;

    v_UTRAN_PhysChId_Cell5 := v_UTRANSysInfo.Utran[0].FDD.ScramblingCode;

     v_HandoverToUtranCommand := f_U_EncodeHandoverToUtran_Speech_FDD(v_UTRANSysInfoFDD.UplinkScrambingCode,

                                                                       v_UTRANSysInfoFDD.ScramblingCode,

                                                                       v_UTRANSysInfoFDD.Freq);

    }

    else {

      v_UTRANSysInfoTDD := v_UTRANSysInfo.Utran[0].TDD;

       v_UTRAN_ARFCN_Cell5 := v_UTRANSysInfo.Utran[0].TDD.Freq;

       v_UTRAN_PhysChId_Cell5 := v_UTRANSysInfo.Utran[0].TDD.CellParametersID;

      v_HandoverToUtranCommand := f_U_EncodeHandoverToUtran_Speech_TDD(v_UTRANSysInfoTDD.CellParametersID,

                                                                       v_UTRANSysInfoTDD.Freq);

    }

    //  @siclog "Step 1" siclog@

    //  The SS configures UTRA cell 5 to reference configuration according 36.508 table 4.8.3-1, condition UTRA PS RB + Speech .

    f_EUTRA_SendTriggerToOtherRAT (true);

    //   @siclog "Steps 2-27" siclog@

    // Make the UE attempt an IMS speech call

    f_EUTRA_GenericIMS_MT_SpeechCall(eutra_Cell1);

    //  @siclog "Step 28" siclog@

    //  The SS transmits an RRCConnectionReconfiguration message on Cell 1 to setup inter RAT measurement and reporting for event B2.

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell1,

                                   cs_TimingInfo_Now,

                                   cs_RRCConnectionReconfiguration_SRVCC_FDD(v_RRC_TI,

                                                                               v_Frequency_IE_Cell1.UL_DL_Earfcn.dl_CarrierFreq,

                                                                               v_ChBandDependency_Cell1.AllowedMeasBandwidth,

                                                                               v_UTRAN_ARFCN_Cell5,

                                                                               v_UTRAN_PhysChId_Cell5)));

    //      @siclog "Step 29" siclog@

    //  The UE transmits an RRCConnectionReconfigurationComplete message on Cell 1.

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete(v_RRC_TI)));

    //  @siclog "Step 30" siclog@

    // The SS changes the power level for Cell 1 and Cell 5 according to the row "T1" in table 13.4.3.2.3.2-1

    f_IRAT_SendCoOrd(UTRAN, cms_IRAT_Trigger);

    f_EUTRA_SetCellPower(eutra_Cell1, v_T1_RS_EPRE_Cell1);

    //   @siclog "Step 31" siclog@

    //  The UE transmits a MeasurementReport message on Cell 1 to report event B2 for Cell 5.

    //v_MeasRpt := cr_MeasurementReport_Utran_1Entry_FDD(1, v_UTRAN_PhysChId_Cell5);

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_MeasurementReport_Utran_1Entry_FDD(1, v_UTRAN_PhysChId_Cell5)));
        //   @siclog "Step 32" siclog@

    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellA,

                               cs_TimingInfo_Now,

                               cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                             cs_Notification(tsc_EpsDefaultBearerId,

                                                              tsc_PTI_1)))); 
    //   @siclog "Step 33" siclog@

    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]);

    //Wait Trigger from IMS PTC

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);

  // Release SRBs and DRBs

   f_EUTRA_SS_SRBs_DRBs_Release(eutra_Cell1, cs_TimingInfo_Now);
  } // function fl_TC_13_4_3_6_Body



After:
	      function fl_TC_13_4_3_6_Body() runs on EUTRA_PTC

  {

    var AbsoluteCellPower_Type v_T1_RS_EPRE_Cell1 := -84;

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var integer v_UTRAN_PhysChId_Cell5;

    var octetstring v_HandoverToUtranCommand;

    //var integer v_UTRAN_ARFCN_f8; //Cell5

    var Frequency_IE_Type   v_Frequency_IE_Cell1 := f_EUTRA_CellInfo_GetFrequencyIEs (eutra_Cell1);

    var ChannelBandwidthDependency_Type  v_ChBandDependency_Cell1:= f_EUTRA_BandDependentParam(v_Frequency_IE_Cell1.DL_ChBandwidth, v_Frequency_IE_Cell1.UL_ChBandwidth);

    var integer v_UTRAN_ARFCN_Cell5;

    var IRAT_CoOrd_SysInfo_Type v_UTRANSysInfo;

    var CoOrd_UTRANSysInfoFDD_Type v_UTRANSysInfoFDD;

    var CoOrd_UTRANSysInfoTDD_Type v_UTRANSysInfoTDD;

    var ProcedureTransactionIdentifier v_EpsProcedureTransactionId := '00'O;
    v_UTRANSysInfo := f_IRAT_WaitForCoOrd_SysInfo (UTRAN);

    if (ischosen (v_UTRANSysInfo.Utran[0].FDD)) {

      v_UTRANSysInfoFDD := v_UTRANSysInfo.Utran[0].FDD;

        v_UTRAN_ARFCN_Cell5 := v_UTRANSysInfo.Utran[0].FDD.Freq;

    v_UTRAN_PhysChId_Cell5 := v_UTRANSysInfo.Utran[0].FDD.ScramblingCode;

     v_HandoverToUtranCommand := f_U_EncodeHandoverToUtran_Speech_FDD(v_UTRANSysInfoFDD.UplinkScrambingCode,

                                                                       v_UTRANSysInfoFDD.ScramblingCode,

                                                                       v_UTRANSysInfoFDD.Freq);

    }

    else {

      v_UTRANSysInfoTDD := v_UTRANSysInfo.Utran[0].TDD;

       v_UTRAN_ARFCN_Cell5 := v_UTRANSysInfo.Utran[0].TDD.Freq;

       v_UTRAN_PhysChId_Cell5 := v_UTRANSysInfo.Utran[0].TDD.CellParametersID;

      v_HandoverToUtranCommand := f_U_EncodeHandoverToUtran_Speech_TDD(v_UTRANSysInfoTDD.CellParametersID,

                                                                       v_UTRANSysInfoTDD.Freq);

    }

    //  @siclog "Step 1" siclog@

    //  The SS configures UTRA cell 5 to reference configuration according 36.508 table 4.8.3-1, condition UTRA PS RB + Speech .

    f_EUTRA_SendTriggerToOtherRAT (false);
    //   @siclog "Steps 2-27" siclog@

    // Make the UE attempt an IMS speech call

    f_EUTRA_GenericIMS_MT_SpeechCall(eutra_Cell1);

    //  @siclog "Step 28" siclog@

    //  The SS transmits an RRCConnectionReconfiguration message on Cell 1 to setup inter RAT measurement and reporting for event B2.

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell1,

                                   cs_TimingInfo_Now,

                                   cs_RRCConnectionReconfiguration_SRVCC_FDD(v_RRC_TI,

                                                                               v_Frequency_IE_Cell1.UL_DL_Earfcn.dl_CarrierFreq,

                                                                               v_ChBandDependency_Cell1.AllowedMeasBandwidth,

                                                                               v_UTRAN_ARFCN_Cell5,

                                                                               v_UTRAN_PhysChId_Cell5)));

    //      @siclog "Step 29" siclog@

    //  The UE transmits an RRCConnectionReconfigurationComplete message on Cell 1.

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete(v_RRC_TI)));

    //  @siclog "Step 30" siclog@

    // The SS changes the power level for Cell 1 and Cell 5 according to the row "T1" in table 13.4.3.2.3.2-1

    f_IRAT_SendCoOrd(UTRAN, cms_IRAT_Trigger);

    f_EUTRA_SetCellPower(eutra_Cell1, v_T1_RS_EPRE_Cell1);

    //   @siclog "Step 31" siclog@

    //  The UE transmits a MeasurementReport message on Cell 1 to report event B2 for Cell 5.

    //v_MeasRpt := cr_MeasurementReport_Utran_1Entry_FDD(1, v_UTRAN_PhysChId_Cell5);

     SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_MeasurementReport_Utran_1Entry_FDD_RSCP(1, v_UTRAN_PhysChId_Cell5))); 

    //   @siclog "Step 32" siclog@

    SRB.send(cas_SRB1_NasPdu_REQ(eutra_Cell1,
                               cs_TimingInfo_Now,

                               cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                             cs_Notification(tsc_EpsDefaultBearerId,

                                                              v_EpsProcedureTransactionId)))); 

    //   @siclog "Step 33" siclog@

    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]);

    //Wait Trigger from IMS PTC

    [] IMS[tsc_Index_PDN1].receive(cmr_IMS_IPCAN_Trigger) {}

  } // function fl_TC_13_4_3_6_Body


4.3 Change 3
	Testcase name
	13.4.3.6

	Protocol
	Multilayer

	Function Name
	f_TC_13_4_3_6_UTRAN()

	Reason for change
	1. It is not required to send any coordination trigger back to EUTRA PTC after setting T1 power level

2. The UTRAN postamble sequence is not required as the test case will end with the UE on EUTRAN. Instead UTRAN cell resources should be released at the end of the test cases.

	Summary of change
	1. Remove unnecessary trigger message

2. Remove postamble function and replace with cell release function.

	Source of change
	SRVCC_EUTRA_UTRAN.ttcn


Before:
	  function f_TC_13_4_3_6_UTRAN() runs on UTRAN_PTC

  {
 …………
    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    //@siclog "Step 30" siclog@

    // Set Cell Power level as per T1

    f_UTRAN_SetCellPower(utran_Cell5,  v_T1_CPICH_Cell5,v_T1_PCCPCH_Cell5);

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger); // Tell SS to transmit a MobilityFromEUTRACommand message
    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_UTRAN_TestBody_Set(false);

    // Switch off UE and release cell

    f_UTRAN_Postamble(utran_Cell5, U2_CONNECTED);
  }



After:
	  function f_TC_13_4_3_6_UTRAN() runs on UTRAN_PTC

  {
 …………
    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    //@siclog "Step 30" siclog@

    // Set Cell Power level as per T1

    f_UTRAN_SetCellPower(utran_Cell5,  v_T1_CPICH_Cell5,v_T1_PCCPCH_Cell5);

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_UTRAN_TestBody_Set(false);

    // release cell

    f_UTRAN_ReleaseCell ( utran_Cell5, cell_DCH_StandAloneSRB_NoConn );
  }



4.4 Change 4
	Testcase name
	13.4.3.6

	Protocol
	Multilayer

	Template Name
	cs_RRCConnectionReconfiguration_SRVCC_FDD

	Reason for change
	The current measurement configuration for event B2 has reporting quantity “infinity”. Since no handover to UTRAN happens in this test case, the UE can keep sending additional measurement reports after step 31 as the event B2 conditions are still satisified.

	Summary of change
	Change the reporting quantity to 1 to avoid having to handle additional measurement reports. This is also in line with 36.508 tabel 4.6.6-8.

	Source of change
	EUTRA_Common_SRVCC.ttcn


Before:
	template (value) DL_DCCH_Message cs_RRCConnectionReconfiguration_SRVCC_FDD(RRC_TransactionIdentifier p_RRC_TI,

                                                                             ARFCN_ValueEUTRA p_EUTRA_DL_CarrierFreq1,

                                                                             AllowedMeasBandwidth p_MeasurementBandwidth1,

                                                                             ARFCN_ValueUTRA p_CarrierFreq,

                                                                             PhysCellIdUTRA_FDD p_PhysCellIdUTRA_FDD) :=

  /* @status    APPROVED (LTE_IRAT) */

    cs_508_RRCConnectionReconfiguration_ConditionMEAS(p_RRC_TI,

                                                      cs_MeasConfig(omit,

                                                                    cs_MeasObjectToAddModList_2Entries(1,

                                                                                                       cs_MeasObject_measObjectEUTRA_Common(p_EUTRA_DL_CarrierFreq1, p_MeasurementBandwidth1),

                                                                                                       tsc_IdMeasObject_f8,

                                                                                                       cs_MeasObject_measObjectUTRA(p_CarrierFreq,

                                                                                                                                    0,

                                                                                                                                    omit,

                                                                                                                                    cs_MeasObjectUTRA_cellsToAddModList_FDD(cs_CellsToAddModListUTRA_FDD_1Entry(1, p_PhysCellIdUTRA_FDD)),

                                                                                                                                    omit)),

                                                                    omit,

                                                                    cs_ReportConfigToAddModList_1Entry(1, cs_ReportConfig_reportConfigInterRAT_eventB2_UTRA_RSCP(-72, -76, 3, ms0, 6, ms1024, infinity_)),

                                                                    omit,

                                                                    cs_MeasIdToAddModList_1Entry(1, tsc_IdMeasObject_f8, 1),

                                                                    cs_QuantityConfig(cs_QuantityConfigEUTRA(omit, omit),

                                                                                      cs_QuantityConfigUTRA(cpich_RSCP, pccpch_RSCP, fc3),

                                                                                      omit,

                                                                                      omit),

                                                                    cs_508_MeasGapConfig_GP1));


After:
	template (value) DL_DCCH_Message cs_RRCConnectionReconfiguration_SRVCC_FDD(RRC_TransactionIdentifier p_RRC_TI,

                                                                             ARFCN_ValueEUTRA p_EUTRA_DL_CarrierFreq1,

                                                                             AllowedMeasBandwidth p_MeasurementBandwidth1,

                                                                             ARFCN_ValueUTRA p_CarrierFreq,

                                                                             PhysCellIdUTRA_FDD p_PhysCellIdUTRA_FDD) :=

  /* @status    APPROVED (LTE_IRAT) */

    cs_508_RRCConnectionReconfiguration_ConditionMEAS(p_RRC_TI,

                                                      cs_MeasConfig(omit,

                                                                    cs_MeasObjectToAddModList_2Entries(1,

                                                                                                       cs_MeasObject_measObjectEUTRA_Common(p_EUTRA_DL_CarrierFreq1, p_MeasurementBandwidth1),

                                                                                                       tsc_IdMeasObject_f8,

                                                                                                       cs_MeasObject_measObjectUTRA(p_CarrierFreq,

                                                                                                                                    0,

                                                                                                                                    omit,

                                                                                                                                    cs_MeasObjectUTRA_cellsToAddModList_FDD(cs_CellsToAddModListUTRA_FDD_1Entry(1, p_PhysCellIdUTRA_FDD)),

                                                                                                                                    omit)),

                                                                    omit,

                                                                    cs_ReportConfigToAddModList_1Entry(1, cs_ReportConfig_reportConfigInterRAT_eventB2_UTRA_RSCP(-72, -76, 3, ms0, 6, ms1024, 1)),
                                                                    omit,

                                                                    cs_MeasIdToAddModList_1Entry(1, tsc_IdMeasObject_f8, 1),

                                                                    cs_QuantityConfig(cs_QuantityConfigEUTRA(omit, omit),

                                                                                      cs_QuantityConfigUTRA(cpich_RSCP, pccpch_RSCP, fc3),

                                                                                      omit,

                                                                                      omit),

                                                                    cs_508_MeasGapConfig_GP1));


4.5 Change 5
	Testcase name
	13.4.3.6

	Protocol
	Multilayer

	Function Name
	f_IMS_SRVCC_Call_ReInvite()

	Reason for change
	1. Current implementation calls function f_IMS_MOCallSetup_AnnexC21_Step2_A4() which will not use the correct condition for checking the re-INVITE message.
2. The 200 OK message should include an SDP message body to answer the offer received in the re-INVITE message

3. The received ACK message will have a request URI set to the contact URI sent back to the UE in the 200 OK, not the contact URI from the INVITE.

4. Since the IMS call was initially set up in a mobile-terminated direction, the ACK request should use the route-set defined for an MT call  

Note : A prose CR will be raised on 36.523-1 to add 200 OK and ACK to the expected message sequence for this test case (and also 13.4.3.4)


	Summary of change
	1. Call new function f_IMS_MOCallSetup_AnnexC21_A5()
2. Replay SDP content received in re-INVITE in 200 OK response

3. Use px_CalleeContactURI for expected request URI of ACK
4. Set the expected route header in the ACK request from the function f_IMS_RouteSet_MTCall_TX().

	Source of change
	IMS_CommonSRVCC.ttcn


Before:
	  function f_IMS_SRVCC_Call_ReInvite() runs on IMS_PTC

  {

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var  MessageHeader v_MessageHeader_Invite;

    //var template (value) INVITE_Request v_InviteRequest;

    //var template (value) SDP_Message v_Tx_SDP;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var template (value) MessageHeader v_MessageHeader_InviteResponse_200;

    var template (present) MessageHeader v_MessageHeader_Ack;

    // var IMS_DATA_REQ v_IMS_DATA_REQ;

    var INVITE_Request v_InviteRequest;

    //var MessageHeader v_MessageHeader_Response183;

    //var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;
    // Step 1 - Make the UE attempt an IMS speech call

    // Note: MMI action shall be called before entering this function

    // Step 2

    v_IMS_DATA_REQ := f_IMS_MOCallSetup_AnnexC21_step2_A4();

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    v_InviteRequest := v_IMS_DATA_REQ.Request.Invite;

    v_MessageHeader_Invite := v_InviteRequest.msgHeader;

    v_MessageHeader_InviteResponse_200 := f_IMS_OtherResponse_200_MessageHeaderTX(v_MessageHeader_Invite);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_InviteResponse_200)));

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200);

   IMS_Server.receive(car_IMS_Ack_Request(cr_ACK_Request(f_MessageHeader_GetContactSipUrl(v_MessageHeader_Invite), v_MessageHeader_Ack)));

  }//end of f_IMS_SRVCC_Call_ReInvite


After:
	  function f_IMS_SRVCC_Call_ReInvite() runs on IMS_PTC

  {

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var  MessageHeader v_MessageHeader_Invite;

    //var template (value) INVITE_Request v_InviteRequest;

    //var template (value) SDP_Message v_Tx_SDP;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var template (value) MessageHeader v_MessageHeader_InviteResponse_200;

    var template (present) MessageHeader v_MessageHeader_Ack;

    // var IMS_DATA_REQ v_IMS_DATA_REQ;

    var INVITE_Request v_InviteRequest;

    //var MessageHeader v_MessageHeader_Response183;

    //var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;
    var template (omit) MessageBody v_MessageBodyTx := omit; 

    var template (omit) ContentType v_ContentType := omit;

    var SDP_Message v_SDP_LatestOffer;
    // Step 1 - Make the UE attempt an IMS speech call

    // Note: MMI action shall be called before entering this function

    // Step 2

    v_IMS_DATA_REQ := f_IMS_MOCallSetup_AnnexC21_Step2_A5();
    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    v_InviteRequest := v_IMS_DATA_REQ.Request.Invite;

    v_MessageHeader_Invite := v_InviteRequest.msgHeader;

    // Send 200 OK with SDP Message Body 

    if (ispresent(v_InviteRequest.messageBody)) {

      v_SDP_LatestOffer := v_InviteRequest.messageBody.sdpMessageBody;

      v_ContentType := cs_ContentTypSDP;

      v_MessageBodyTx.sdpMessageBody := f_IMS_BuildSDP_AnnexC21_Step6or8(v_SDP_LatestOffer);

    }
    v_MessageHeader_InviteResponse_200 := f_IMS_OtherResponse_200_MessageHeaderTX(v_MessageHeader_Invite, v_ContentType);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_InviteResponse_200, v_MessageBodyTx)));

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200);

    v_MessageHeader_Ack.route.routeBody := f_IMS_RouteSet_MT_Call_TX();
    IMS_Server.receive(car_IMS_Ack_Request(cr_ACK_Request(f_SIP_BuildSipUri_RX(px_IMS_CalleeContactUri), v_MessageHeader_Ack)));

  }//end of f_IMS_SRVCC_Call_ReInvite


4.6 Change 6
	Testcase name
	13.4.3.6

	Protocol
	Multilayer

	Function Name
	f_IMS_MOCallSetup_AnnexC21_A5()

	Reason for change
	No function currently exists to initiate checking of re-INVITE request within an existing dialog, corresponding to condition A5 in 34.229 Annex A.2.1

	Summary of change
	Create new function f_IMS_MOCallSetup_AnnexC21_Step2_A5() to call f_IMS_MOCallSetup_AnnexC21_Step2_Common() indicating condition A5.

	Source of change
	IMS_Procedures_CallControl.ttcn


New function:
	  function f_IMS_MOCallSetup_AnnexC21_Step2_A5(boolean p_GeolocationInfoAvailable := false) runs on IMS_PTC return IMS_DATA_REQ

  { /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    return f_IMS_MOCallSetup_AnnexC21_Step2_Common(-, A_2_1_A5, p_GeolocationInfoAvailable);

  }


4.7 Change 7
	Testcase name
	13.4.3.6

	Protocol
	Multilayer

	Function Name
	f_IMS_MOCallSetup_AnnexC21_Step2_Optional

	Reason for change
	1. Since for condition A5 the IMS call is already setup and resource reservation is completed for, it is not necessary for the UE to include the “precondition” option tag in the re-INVITE.
2. The function f_IMS_BuildSDP_AnnexC21_Step2() will create the expected initial SDP offer from the UE in the case of an MO call. In the case of condition A5 all SDP negotiation has been completed so this will not match the received content.

	Summary of change
	1. Allow the UE to not include the “precondition” option tag in the case of condition A5

2. Proposed implementation is to allow any SDP content in the re-INVITE message, since the precise contents are not mandated by the test case prose.

	Source of change
	IMS_CommonProcedures_CallControl.ttcn


Before:
	  function f_IMS_MOCallSetup_AnnexC21_Step2_Optional(template (omit) charstring p_OtherCalleeUri := omit,

                                                     INVITE_A_2_1_Context_Type p_A_2_1_Context := A_2_1_A4,

                                                     template (omit) float p_WaitDuration := omit,

                                                     boolean p_GeolocationInfoAvailable := false) runs on IMS_PTC return template (omit) IMS_DATA_REQ

  { /* @sic R5-133276, R5-133301 Cleanup of functions for MO Call Setup sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    var template (present) SDP_Message v_SDP_Message_Step2 := f_IMS_BuildSDP_AnnexC21_Step2();

    var template (present) MessageBody v_MessageBody_Step2;

    var template (omit) IMS_DATA_REQ v_IMS_DATA_REQ;

    var IMS_DATA_REQ v_IMS_DATA_REQ_Value;

    var SDP_Message v_SDP_Message;

    var template(value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var INVITE_Request v_InviteRequest;

    v_MessageBody_Step2 := f_SIP_MessageBodySDP_RX(v_SDP_Message_Step2, p_GeolocationInfoAvailable);

    // Step 2. Receive INVITE Request

    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest_Common(p_A_2_1_Context, tsc_OptionTagList_precondition, v_MessageBody_Step2, p_OtherCalleeUri, p_WaitDuration, -, p_GeolocationInfoAvailable);

    if (isvalue(v_IMS_DATA_REQ)) {

      v_IMS_DATA_REQ_Value := valueof(v_IMS_DATA_REQ);

      v_InviteRequest := v_IMS_DATA_REQ_Value.Request.Invite;

      v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ_Value.RoutingInfo);

      v_SDP_Message := f_IMS_MessageBody_CheckGeolocationAndGetSDP(v_InviteRequest.messageBody, p_GeolocationInfoAvailable);

      // check at least one connections field sent

      f_IMS_SDP_CheckConnections(v_SDP_Message);

      // check media description:

      f_IMS_SDP_CheckMediaDescr_AMR_8000_1(v_SDP_Message.media_list[0]);

      f_IMS_SDP_CheckMediaDescr_DTMF(v_SDP_Message.media_list[0]);

    }

    return v_IMS_DATA_REQ;

  }


After:
	  function f_IMS_MOCallSetup_AnnexC21_Step2_Optional(template (omit) charstring p_OtherCalleeUri := omit,

                                                     INVITE_A_2_1_Context_Type p_A_2_1_Context := A_2_1_A4,

                                                     template (omit) float p_WaitDuration := omit,

                                                     boolean p_GeolocationInfoAvailable := false) runs on IMS_PTC return template (omit) IMS_DATA_REQ

  { /* @sic R5-133276, R5-133301 Cleanup of functions for MO Call Setup sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    var template (present) SDP_Message v_SDP_Message_Step2 := f_IMS_BuildSDP_AnnexC21_Step2();

    var template (present) MessageBody v_MessageBody_Step2;

    var template (omit) IMS_DATA_REQ v_IMS_DATA_REQ;

    var IMS_DATA_REQ v_IMS_DATA_REQ_Value;

    var SDP_Message v_SDP_Message;

    var template(value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var INVITE_Request v_InviteRequest;

    var OptionTag_List v_AdditionalOptions := tsc_OptionTagList_Empty;

   if (p_A_2_1_Context != A_2_1_A5)

    {

       v_SDP_Message_Step2 := f_IMS_BuildSDP_AnnexC21_Step2();

       v_AdditionalOptions := tsc_OptionTagList_precondition;

    }

    else

    {

       v_SDP_Message_Step2 := cr_SDP_Message_AnySession(f_IMS_PTC_NW_Address_GetTypeStr(), ?, ?, ?);
    }
    v_MessageBody_Step2 := f_SIP_MessageBodySDP_RX(v_SDP_Message_Step2, p_GeolocationInfoAvailable);

    // Step 2. Receive INVITE Request

    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest_Common(p_A_2_1_Context, v_AdditionalOptions, v_MessageBody_Step2, p_OtherCalleeUri, p_WaitDuration, -, p_GeolocationInfoAvailable);

    if (isvalue(v_IMS_DATA_REQ)) {

      v_IMS_DATA_REQ_Value := valueof(v_IMS_DATA_REQ);

      v_InviteRequest := v_IMS_DATA_REQ_Value.Request.Invite;

      v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ_Value.RoutingInfo);

      v_SDP_Message := f_IMS_MessageBody_CheckGeolocationAndGetSDP(v_InviteRequest.messageBody, p_GeolocationInfoAvailable);

      if (p_A_2_1_Context != A_2_1_A5)

      {
      // check at least one connections field sent

      f_IMS_SDP_CheckConnections(v_SDP_Message);

      // check media description:

      f_IMS_SDP_CheckMediaDescr_AMR_8000_1(v_SDP_Message.media_list[0]);

      f_IMS_SDP_CheckMediaDescr_DTMF(v_SDP_Message.media_list[0]);

     }

    }
    return v_IMS_DATA_REQ;

  }  


4.8 Change 8
	Testcase name
	13.4.3.6

	Protocol
	Multilayer

	Function Name
	f_IMS_INVITE_ReceiveRequest_Common

	Reason for change
	1. For condition A5 (re-INVITE within an existing dialog), the re-INVITE message will will have a request URI equivalent to px_IMS_CalleeContactUri rather than px_IMS_CalleeUri. Also since the dialog is already established, a host port may be included.
2. The IMS PTS globacl dialog info store should not be reinitialised in the case of condition A5 (re-INVITE within an existing dialog). However since the re-INVITE is a request initiated by the UE, the to-tag may change and should be saved for use in future requests and responses. 

	Summary of change
	1. Set expected calleeUri to the value of px_CalleeContactUri for condition A5, and allow for the inclusion of a hostport value
2. In the conjdition A5, avoid calling f_IMS_PTC_ImsInfo_DialogInit(). Instead store the reiceved to-tag.

	Source of change
	IMS_CommonProcedures_CallControl.ttcn


Before:
	  function f_IMS_INVITE_ReceiveRequest_Common(INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                              OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                              template (present) MessageBody p_MessageBody,

                                              template (omit) charstring p_OtherCalleeUri := omit,

                                              template (omit) float p_WaitDuration := omit,

                                              boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                              boolean p_A8 := pc_IMS_Geolocation,

                                              boolean p_A9 := false,

                                              boolean p_A10 := false,

                                              boolean p_A11 := false,

                                              boolean p_A12 := false)

    runs on IMS_PTC return template (omit) IMS_DATA_REQ

  { /* p_OtherCalleeUri .. if present to be used as CalleeUri; if omit default CalleeUri is used acc. to PIXITs and cntext */

    /* @sic R5-133276, R5-133301 Cleanup of functions for MO Call Setup sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (present) MessageHeader v_MessageHeader_Invite;

    var template (present) SipUrl v_ExpectedCalleeUri;

    var boolean v_TimerControl := ispresent(p_WaitDuration);

    timer t_Timer;

    if (ispresent(p_OtherCalleeUri)) {

      v_ExpectedCalleeUri := f_SIP_BuildSipUri_RX(valueof(p_OtherCalleeUri));

    } else {

      select (p_Context) {

        case (A_2_1_A4) {                // A4: INVITE creating a dialog

          v_ExpectedCalleeUri := f_SIP_BuildSipUri_RX(px_IMS_CalleeUri);

        }

        case (A_2_1_A5) {                // A5: re-INVITE

          // v_ExpectedCalleeUri := f_SIP_BuildSipUri_RX(px_IMS_CalleeContactUri);
        }

        case (A_2_1_A6, A_2_1_A7) {      // A6, A7: emergency call

          v_ExpectedCalleeUri := cr_SipUrl_Common(c_urnScheme, cr_SipUrl_SOS);

        }

        case else {

          FatalError(__FILE__, __LINE__, "");

        }

      }

    }

    v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderRX(v_ExpectedCalleeUri, p_AdditionalOptionTagsForSupported, p_Context, p_A3, p_A8, p_A9, p_A11);

    if (v_TimerControl) {

      t_Timer.start(valueof(p_WaitDuration));

    }

    alt {

      [] IMS_Server.receive(car_IMS_Invite_Request(cr_INVITE_Request(v_ExpectedCalleeUri, v_MessageHeader_Invite, p_MessageBody))) -> value v_IMS_DATA_REQ

        {

          t_Timer.stop;

          if (p_Context == A_2_1_A6) {      // emergency call without registration   @sic R5s130756 change 14 sic@

            f_IMS_PTC_ImsInfo_Init(v_IMS_DATA_REQ.RoutingInfo.UE_Address, v_IMS_DATA_REQ.RoutingInfo.NW_Address);

          }

          f_IMS_A_2_1_Invite_CommonChecks(v_IMS_DATA_REQ, p_Context, p_A3, p_A8, p_A10, p_A11, p_A12);

          f_IMS_PTC_ImsInfo_DialogInit(dialogRemote, v_IMS_DATA_REQ.Request.Invite.msgHeader);
          return v_IMS_DATA_REQ;

        }

      [v_TimerControl] t_Timer.timeout {}

    }

    return omit;

  }


After:
	    function f_IMS_INVITE_ReceiveRequest_Common(INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                              OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                              template (present) MessageBody p_MessageBody,

                                              template (omit) charstring p_OtherCalleeUri := omit,

                                              template (omit) float p_WaitDuration := omit,

                                              boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                              boolean p_A8 := pc_IMS_Geolocation,

                                              boolean p_A9 := false,

                                              boolean p_A10 := false,

                                              boolean p_A11 := false,

                                              boolean p_A12 := false)

    runs on IMS_PTC return template (omit) IMS_DATA_REQ

  { /* p_OtherCalleeUri .. if present to be used as CalleeUri; if omit default CalleeUri is used acc. to PIXITs and cntext */

    /* @sic R5-133276, R5-133301 Cleanup of functions for MO Call Setup sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (present) MessageHeader v_MessageHeader_Invite;

    var template (present) SipUrl v_ExpectedCalleeUri;

    var boolean v_TimerControl := ispresent(p_WaitDuration);

    var template (omit) charstring v_ReceivedToTag;
    timer t_Timer;

    if (ispresent(p_OtherCalleeUri)) {

      v_ExpectedCalleeUri := f_SIP_BuildSipUri_RX(valueof(p_OtherCalleeUri));

    } else {

      select (p_Context) {

        case (A_2_1_A4) {                // A4: INVITE creating a dialog

          v_ExpectedCalleeUri := f_SIP_BuildSipUri_RX(px_IMS_CalleeUri);

        }

        case (A_2_1_A5) {                // A5: re-INVITE

          v_ExpectedCalleeUri := f_SIP_BuildSipUri_RX(px_IMS_CalleeContactUri, *);
        }

        case (A_2_1_A6, A_2_1_A7) {      // A6, A7: emergency call

          v_ExpectedCalleeUri := cr_SipUrl_Common(c_urnScheme, cr_SipUrl_SOS);

        }

        case else {

          FatalError(__FILE__, __LINE__, "");

        }

      }

    }

    v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderRX(v_ExpectedCalleeUri, p_AdditionalOptionTagsForSupported, p_Context, p_A3, p_A8, p_A9, p_A11);

    if (v_TimerControl) {

      t_Timer.start(valueof(p_WaitDuration));

    }

    alt {

      [] IMS_Server.receive(car_IMS_Invite_Request(cr_INVITE_Request(v_ExpectedCalleeUri, v_MessageHeader_Invite, p_MessageBody))) -> value v_IMS_DATA_REQ

        {

          t_Timer.stop;

          if (p_Context == A_2_1_A6) {      // emergency call without registration   @sic R5s130756 change 14 sic@

            f_IMS_PTC_ImsInfo_Init(v_IMS_DATA_REQ.RoutingInfo.UE_Address, v_IMS_DATA_REQ.RoutingInfo.NW_Address);

          }

          f_IMS_A_2_1_Invite_CommonChecks(v_IMS_DATA_REQ, p_Context, p_A3, p_A8, p_A10, p_A11, p_A12);

          //f_IMS_PTC_ImsInfo_DialogInit(dialogRemote, v_IMS_DATA_REQ.Request.Invite.msgHeader);

          if (p_Context != A_2_1_A5) 

          {

              f_IMS_PTC_ImsInfo_DialogInit(dialogRemote, v_IMS_DATA_REQ.Request.Invite.msgHeader);

          }

          else

          {

              v_ReceivedToTag := f_MessageHeader_GetToTag(v_IMS_DATA_REQ.Request.Invite.msgHeader);// store the new to-tag

              f_IMS_PTC_ImsInfo_DialogSetToTag(v_ReceivedToTag);

          }
          return v_IMS_DATA_REQ;

        }

      [v_TimerControl] t_Timer.timeout {}

    }

    return omit;

  }


4.9 Change 9
	Testcase name
	13.4.3.6

	Protocol
	Multilayer

	Function Name
	F_IMS_InviteRequest_MessageHeaderRX

	Reason for change
	1. Since the IMS call was initially set up in a mobile-terminated direction, the re-INVITE request should use the route-set defined for an MT call  

2. Since a dialog has been established, the re-INVITE request will contain a tag in the “to” header

3. Since for condition A5 the INIVITE request is not establishing a new dialog, the checking several headers (including “accept”, “proxeRequire”, “require”, “acceptContact” and “pPreferredService”) is not required.

	Summary of change
	1. For condition A5, use the function f_IMS_RouteSet_MT_Call_TX() to set the expected route header content
2. For condition A5, expect a tag in the “to” header

3. Allow some headers (including “accept”, “proxeRequire”, “require”, “acceptContact” and “pPreferredService”) to be omitted in the case of consition A5

	Source of change
	IMS_CommonProcedures_CallControl.ttcn


Before:
	    function f_IMS_InviteRequest_MessageHeaderRX(template (present) SipUrl p_ExpectedCalleeUri,

                                               OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                               INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                               boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                               boolean p_A8 := pc_IMS_Geolocation,

                                               boolean p_A9 := false,

                                               boolean p_A11 := false)

    runs on IMS_PTC return template (present) MessageHeader

  {
 ……………
    // route body acc. to conditions in A.2.1

    v_RouteBodyList[0] := cr_RouteBody(*, cr_SipUri_HostPort_lr(f_IMS_PTC_Pcscf_Get(), v_Port_ps));

    if (not match(p_Context, (A_2_1_A6, A_2_1_A7))) {  // => no emergency call

      v_RouteBodyList[1] := cr_RouteBody(*, cr_SipUri_HostPort_lr(px_IMS_Scscf));

    }

    if (p_A8) {

      v_MessageHeader_Invite.geolocation := cr_GeolocationAny;

      v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingSupported; // @sic R5-134798 sic@

      v_MediaType := "multipart/mixed";

    }

    v_MessageHeader_Invite.via             := fl_IMS_InviteRequest_ViaRX(v_IsGiba);                   /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address

                                                                                                         @sic R5-133358: rport is optional for tcp sic@ */

    v_MessageHeader_Invite.route           := cr_Route(v_RouteBodyList);

    v_MessageHeader_Invite.fromField       := cr_From(v_Displayname, v_SipUrlFrom);                   /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address, check against P-Preferred-Identity header */

    v_MessageHeader_Invite.toField         := cr_ToDef(p_ExpectedCalleeUri);                          /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of tag */

    v_MessageHeader_Invite.callId          := cr_CallId(?);                                           /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: value acc. to A4 or A5 */

    v_MessageHeader_Invite.sessionId       := f_IMS_SessionId_RX_IfSupported(p_A9);                   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Invite.cSeq            := cr_CseqDef(?, "INVITE");                                /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: A5 */

    v_MessageHeader_Invite.supported       := cr_Supported(v_OptionTag_List);

    v_MessageHeader_Invite.contact         := cr_Contact(cr_SipUri_HostPort(?, v_Port_us));           /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address, check of feature-param for A3, A6, A10, A12 */

    v_MessageHeader_Invite.contentType     := cr_ContentType(v_MediaType);

    v_MessageHeader_Invite.maxForwards     := cr_MaxForwardsDef;

// additional headers for specific conditions:

……..
    if (p_A3) {

      v_MessageHeader_Invite.pPreferredService := cr_PPreferredService(tsc_IMS_AcceptContactValue);

    }

    if (p_A3 or p_A11) {

      v_MessageHeader_Invite.acceptContact   := cr_AcceptContact;    // Checked in f_IMS_A_2_1_Invite_CommonChecks (depending on conditions)

    }

    return v_MessageHeader_Invite;

  }



After:
	  function f_IMS_InviteRequest_MessageHeaderRX(template (present) SipUrl p_ExpectedCalleeUri,

                                               OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                               INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                               boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                               boolean p_A8 := pc_IMS_Geolocation,

                                               boolean p_A9 := false,

                                               boolean p_A11 := false)

    runs on IMS_PTC return template (present) MessageHeader

  {
 …………
    // route body acc. to conditions in A.2.1

    v_RouteBodyList[0] := cr_RouteBody(*, cr_SipUri_HostPort_lr(f_IMS_PTC_Pcscf_Get(), v_Port_ps));

    if (not match(p_Context, (A_2_1_A5, A_2_1_A6, A_2_1_A7))) {  // => no emergency call

      v_RouteBodyList[1] := cr_RouteBody(*, cr_SipUri_HostPort_lr(px_IMS_Scscf));

    }

     if (p_Context == A_2_1_A5)
    {

        v_RouteBodyList := f_IMS_RouteSet_MT_Call_TX();

    }
    if (p_A8) {

      v_MessageHeader_Invite.geolocation := cr_GeolocationAny;

      v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingSupported; // @sic R5-134798 sic@

      v_MediaType := "multipart/mixed";

    }

    v_MessageHeader_Invite.via             := fl_IMS_InviteRequest_ViaRX(v_IsGiba);                   /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address

                                                                                                         @sic R5-133358: rport is optional for tcp sic@ */

    v_MessageHeader_Invite.route           := cr_Route(v_RouteBodyList);

    v_MessageHeader_Invite.fromField       := cr_From(v_Displayname, v_SipUrlFrom);                   /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address, check against P-Preferred-Identity header */

    //v_MessageHeader_Invite.toField         := cr_ToDef(p_ExpectedCalleeUri);                          /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of tag */

    if (not match (p_Context, A_2_1_A5))

    {

        v_MessageHeader_Invite.toField         := cr_ToDef(p_ExpectedCalleeUri);                          /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of tag */

    }

    else

    {

        v_MessageHeader_Invite.toField         := cr_ToDef(p_ExpectedCalleeUri,?);
    }
    v_MessageHeader_Invite.callId          := cr_CallId(?);                                           /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: value acc. to A4 or A5 */

    v_MessageHeader_Invite.sessionId       := f_IMS_SessionId_RX_IfSupported(p_A9);                   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Invite.cSeq            := cr_CseqDef(?, "INVITE");                                /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: A5 */

    v_MessageHeader_Invite.supported       := cr_Supported(v_OptionTag_List);

    v_MessageHeader_Invite.contact         := cr_Contact(cr_SipUri_HostPort(?, v_Port_us));           /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address, check of feature-param for A3, A6, A10, A12 */

    v_MessageHeader_Invite.contentType     := cr_ContentType(v_MediaType);

    v_MessageHeader_Invite.maxForwards     := cr_MaxForwardsDef;

    // additional headers for specific conditions:

………….
    if (p_A3) {

      v_MessageHeader_Invite.pPreferredService := cr_PPreferredService(tsc_IMS_AcceptContactValue);

    }

    if (p_A3 or p_A11) {

      v_MessageHeader_Invite.acceptContact   := cr_AcceptContact;    // Checked in f_IMS_A_2_1_Invite_CommonChecks (depending on conditions)

    }

    if(p_Context == A_2_1_A5){

      v_MessageHeader_Invite.accept          := *;

      v_MessageHeader_Invite.proxyRequire    := *;

      v_MessageHeader_Invite.require         := *;

      v_MessageHeader_Invite.acceptContact   := *; 

      v_MessageHeader_Invite.pPreferredService := *;

    all fields not required if dialog already exists
    return v_MessageHeader_Invite;

  }


4.10 Change 10
	Testcase name
	13.4.3.6

	Protocol
	Multilayer

	Function Name
	f_IMS_A_2_1_Invite_CommonChecks

	Reason for change
	In the current TTCN implementation, when performing checks on a received INVITE message according to 34.229 Annex A.2.1 with condition A5 (re-INVITE within existing dialog), several checks are not implemented and the code generates fatal errors in this case.

	Summary of change
	For this test case verification, the code generating fatal errors has been deactivated. It is still for future study how the required checks on the re-INVITE message are implemented.

	Source of change
	IMS_CommonProceudres_CallControl.ttcn


Before:
	function f_IMS_A_2_1_Invite_CommonChecks(IMS_DATA_REQ p_IMS_INVITE_REQ,

                                           INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                           boolean p_A3 := false,

                                           boolean p_A8 := false,

                                           boolean p_A10 := false,

                                           boolean p_A11 := false,

                                           boolean p_A12 := false) runs on IMS_PTC

  { /* @sic R5-133276, R5-133301 Cleanup of functions for MO Call Setup sic@ */

    ………….

      case (A_2_1_A5) {

        FatalError(__FILE__, __LINE__, "A.2.1 condition A5: Check of Call-ID not implemented yet (value of Call-ID as in any previous request in the same dialog)");

        FatalError(__FILE__, __LINE__, "A.2.1 condition A5: Check of CSeq not implemented yet ()");

        FatalError(__FILE__, __LINE__, "A.2.1 condition A5: From addr-spec shall be checked");

        FatalError(__FILE__, __LINE__, "A.2.1 condition A5: From tag shall be checked");

        FatalError(__FILE__, __LINE__, "A.2.1 condition A5: Request-URI shall be checked");
      }

…..

  }


After:
	  function f_IMS_A_2_1_Invite_CommonChecks(IMS_DATA_REQ p_IMS_INVITE_REQ,

                                           INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                           boolean p_A3 := false,

                                           boolean p_A8 := false,

                                           boolean p_A10 := false,

                                           boolean p_A11 := false,

                                           boolean p_A12 := false) runs on IMS_PTC

  { /* @sic R5-133276, R5-133301 Cleanup of functions for MO Call Setup sic@ */

    ………….

      case (A_2_1_A5) {

        //FatalError(__FILE__, __LINE__, "A.2.1 condition A5: Check of Call-ID not implemented yet (value of Call-ID as in any previous request in the same dialog)");

        //FatalError(__FILE__, __LINE__, "A.2.1 condition A5: Check of CSeq not implemented yet ()");

        //FatalError(__FILE__, __LINE__, "A.2.1 condition A5: From addr-spec shall be checked");

        //FatalError(__FILE__, __LINE__, "A.2.1 condition A5: From tag shall be checked");

        //FatalError(__FILE__, __LINE__, "A.2.1 condition A5: Request-URI shall be checked");
      }

…..

  }


4.11 Change 11
	Testcase name
	13.4.3.6

	Protocol
	Multilayer

	Function Name
	F_SIP_BuildSip_Uri_RX

	Reason for change
	When building a recieve template for a SIP URL, the check “ispresent(p_Port)” does not allow “ * “ to be passed as the port parameter (as has been done in change 8.2)

	Summary of change
	Remove the ispresent(p_Port) check, since it should be allowable for the portField to be set to “*” or “omit”.

	Source of change
	IMS_SIP_Templates.ttcn


Before:
	  function f_SIP_BuildSipUri_RX(charstring p_URI,

                                template integer p_Port := omit,

                                template GenericParam p_LrParam := omit) return template (present) SipUrl

  {

    var template (present) SipUrl v_SipUrl := fl_SIP_ParseURI(p_URI);

    if (ispresent(p_Port) and ischosen(v_SipUrl.components.sip)) { /* @sic R5s130195 BASELINE MOVING 2013: change of SipUrl sic@ */

      v_SipUrl.components.sip.hostPort.portField := p_Port;

    }

    if (ispresent(p_LrParam)) {

      v_SipUrl.urlParameters := {p_LrParam, *};

    } else {

      v_SipUrl.urlParameters := *;

    }

    return v_SipUrl;

  }


After:
	  function f_SIP_BuildSipUri_RX(charstring p_URI,

                                template integer p_Port := omit,

                                template GenericParam p_LrParam := omit) return template (present) SipUrl

  {

    var template (present) SipUrl v_SipUrl := fl_SIP_ParseURI(p_URI);

    if (/*ispresent(p_Port) and */ischosen(v_SipUrl.components.sip)) { /* @sic R5s130195 BASELINE MOVING 2013: change of SipUrl sic@ */

      v_SipUrl.components.sip.hostPort.portField := p_Port;

    }

    if (ispresent(p_LrParam)) {

      v_SipUrl.urlParameters := {p_LrParam, *};

    } else {

      v_SipUrl.urlParameters := *;

    }

    return v_SipUrl;

  }


5. Execution Log Files

Qualcomm MSM9x25 chipset  
The Qualcomm MSM 9x25 chipset UE passed this test case on Anite Conformance Toolset Solution in LTE FDD band 1. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
Anite\TC_13_4_3_6-Log.htm   
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.


References

	[1]
	R5s140021:    Supporting information for agreement of LTE TC 13.4.3.6. This archive comprises:

